Syntaxanalyse (Parsing)

Vorlesung WS 2006/07

Hans Leiß, CIS

16. Januar 2007

(i) Nicht-deterministische Analyse (Backtracking) (a) Nicht-deterministische Top-Down-Analyse (b) Nicht-deterministische Bottom-Up-Analyse (c) Nicht-deterministische Left-Corner-Analyse (d) Nicht-deterministische Head-Corner-Analyse (ii) Tabellengesteuerte Analyse (Chart-Parser) (a) Cocke-Younger-Kasami (b) Einfache Earley-Analyse (c) Einfache Left-Corner-Analyse (d) Vorberechnung von Erreichbarkeitsrelationen (e) Verallgemeinerte Earley-Analyse (f) Verallgemeinerte Left-Corner-Analyse (iii) Deterministische Analyse (mit k-Vorausschau)

(a) Top-Down: LL(k)-Anlayse (b) Bottom-Up: LR(k)-Analyse (c) Left-Corner: LC(k)-Analyse (d) (*) Verallgemeinerte LR(k)-Analyse (Tomita) (iv) Kopfgrammatiken und gespaltene Konstituenten

Phasen der Textanalyse

(i) Lexikalische Analyse:

(a) Reader: Lesen eines Stroms (Datei,Konsole) von Zeichen (Character) (b) Tokenizer: Zerlegen des Stroms in Segmente (Token) (c) Lemmatizer: Annotierung der Token mit Lexikon- Information (d) Tagger: Entfernung irrelevanter Lexikon-Information (ii) Syntaktische Analyse:

(a) Parser: Aufbau der syntaktischen Struktur der getaggten Token-Folge (iii) Semantische Analyse:

(a) Annotierung der syntaktischen Struktur mit semantischen Werten (iv) Übersetzung:

(a) Erzeugung eines Ausgabetexts in der Zielsprache (unter Erhaltung der semantischen Werte)

Lexikalische Analyse

Reader

(i) Falls die Eingabedatei noch nicht zum Lesen geöffnet ist, öffne sie zum Lesen und gib einen Namen für den dabei erzeugten Lesestrom aus.

(ii) Lies vom Eingabestrom bis zum ersten Satzende.

(iii) Gib den gelesenen Satz an den Tokenizer.

(iv) Auf Anforderung lies den nächsten Satz vom Eingabestrom oder schließe den Eingabestrom.

⟨Strom-Prädikate von SWI-Prolog z.B.⟩≡

current_stream(Datei?,Bearb.modus?,Stromname?)
open(Datei+,Bearbeitungsmodus+,Stromname-)

Mit read sentence(File+,Sentence-) erfolgen die ersten beiden Schritte:

⟨Reader 1.+2.⟩≡

read_sentence(File,Sentence) :-
        (current_stream(File,read,Stream)
        -> true
        ; open(File,read,Stream)
        ),
        read_sentence(Stream,Sentence,[]).

Mit read sentence(Strom+,Sentence-,Puffer?) wird das Satzende gesucht.

Satzende-Erkenner

   read sentence(Strom+,Sentence-,Puffer?)
read sentence(Strom+,Sentence-,Char+,Puffer?)

(i) Falls der Eingabestrom zu Ende ist (eof oder ^C), melde das (vorzeitige) Dateiende und brich ab.

(ii) Falls der Eingabestrom noch nicht zu Ende ist, lies ein weiteres Zeichen vom Strom.

(iii) Falls dieses Zeichen eof oder zusammen mit einem beschränkten Linkskontext (auf dem Puffer!) ein Satzende ist, gib die gelesene Zeichenfolge als Sentence zurück.

(iv) Sonst erweitere den Puffer um das gelesene Zeichen und suche weiter nach dem Satzende.

In Prolog sind Strings ASCII-Listen: "abc" = [97, 98, 99], z.B. Sentence und Puffer?.

⟨Satzende-Erkenner, 1.+2.⟩≡

read_sentence(Stream,Sentence,Seen) :-
        get0(Stream,Char),
        !,read_sentence(Stream,Sentence,Char,Seen),
        (dateiende(Char)
         -> write(’End of file reached.’)
        ; true).

⟨Satzende-Erkenner, 3.a⟩≡

read_sentence(_Stream,Sentence,Char,Seen) :-
        dateiende(Char),
        !,reverse(Seen,Sentence).

⟨Sonderfall: Char ist Leerzeichen nach Punkt im Satz ⟩ ⟨Sonderfall: Char ist ein Kommentarzeichen, z.B. %⟩

Satzende-Erkenner (Forts.)

Unter den Sonderfällen muß man (Übung!) behandeln:

• Punkte am Ende von Ordinalzahlen (der 3. Mann)

• Punkte am Ende von Abkürzungen (Dr. Sinn)

Das Satzende satzende([C,Char]) ist erreicht bei ". ".

⟨Satzende-Erkenner, 3.b⟩≡

read_sentence(_Stream,Sentence,Char,[C|Seen]) :-
        satzende([C,Char]),
        !,reverse([C|Seen],Sentence).

⟨Satzende-Erkenner, 4.⟩≡

read_sentence(Stream,Sentence,Char,Seen) :-
        read_sentence(Stream,Sentence,[Char|Seen]).

Statt von einer Datei kann man auch vom Strom user lesen:

⟨Satzende-Test an der Konsole⟩≡

satzende :-
        write(’Beende den Eingabesatz mit
               <Satzzeichen><Return>’),nl,
        read_sentence(user,Sentence,[]),
        name(Satz,Sentence),
        nl,writeq(Satz).

Tokenizer

stringToAtoms(String+,Atomlist-)

Der Satz String muß an den Leerzeichen in Wörter (Token) zerlegt werden. Dazu

(i) Lies den Satz zeichenweise, (ii) Stapele gelesene Zeichen als Folge von Token, (iii) Sammle im obersten Stapelelement (invertiertes Präfix eines Tokens!) weitere Character bis zum Tokenende.

(iv) Invertiere den Stapel der Token zu einer Atomliste. Man unterscheidet nach dem zuletzt gelesenen Zeichen: (nt) Das letzte Zeichen beendete ein Token.

tokenize(String+,nt,SeenTokensRev?,Atoms-)

(at) Das letzte Zeichen lag in einem angefangenen Token.

tokenize(String+,at,SeenTokensRev?,Atoms-)

Diese Fälle rufen sich gegenseitig auf, beginnend mit:

⟨Aufruf des Tokenizers⟩≡

stringToAtoms(String,Atomlist) :-
        tokenize(String,nt,[],Atomlist).

Tokenizer (Details 1)

(nt) Das letzte Zeichen beendete ein Token.

⟨Ignoriere führende Leerzeichen und Zeilenumbrüche:⟩≡

tokenize([C|String],nt,Rs,Ts) :-
        (leerZeichen(C); zeilenende(C)),
        !,tokenize(String,nt,Rs,Ts).

⟨Mache Sonderzeichen zu neuen Token:⟩≡

tokenize([C|String],nt,Rs,Ts) :-
        sonderzeichen(Atom,C),
        !,tokenize(String,nt,[Atom|Rs],Ts).

⟨Sonst wird das Zeichen C der Anfang eines Tokens:⟩≡

tokenize([C|String],nt,Rs,Ts) :-
        !,tokenize(String,at,[[C]|Rs],Ts).

⟨Falls der Eingabestring leer ist, gib die Token aus:⟩≡

tokenize([],nt,Rs,Ts) :-
        reverse(Rs,Ts).

(at) Das letzte Zeichen lag in einem angefangenen Token.

⟨Ignoriere Leerzeichen nach Worttrennung am Zeilenende:⟩≡

tokenize([N,L|String],at,[[T|Cs]|Rs],Ts) :-
        worttrennung([T,N]),leerZeichen(L),
        !,tokenize([N|String],at,[[T|Cs]|Rs],Ts).

⟨Ignoriere Worttrennung am Zeilenende:⟩≡

tokenize([N|String],at,[[T|Cs]|Rs],Ts) :-
        worttrennung([T,N]),
        !,tokenize(String,at,[Cs|Rs],Ts).

Tokenizer (Details 2)

⟨Am Zeilenende ohne Worttrenner endet ein Token:⟩≡

tokenize([C|String],at,[R|Rs],Ts) :-
        zeilenende(C),
        !,tokenToAtom(R,Atom),
        tokenize(String,nt,[Atom|Rs],Ts).

Die gesammelten Zeichen des Tokens werden in die normale Reihenfolge gebracht und zu einem Atom gemacht:

⟨Umwandlung der Zeichenfolge des Tokens in ein Atom:⟩≡

tokenToAtom(RevToken,Atom) :-
        reverse(RevToken,Token),
        name(Atom,Token).

Das Atom kommt auf die Liste der Token und man sucht wieder nach einem neuen Token (Fall nt).

⟨Weitere zu behandelnde Fälle⟩≡

⟨An einem Leerzeichen endet ein Token⟩ ⟨Ziffer, Punkt/Komma, Leerzeichen beenden das Token⟩ ⟨Ziffer, Punkt/Komma, Ziffer setzen das Token fort⟩ ⟨Punkt am Ende einer Abkürzung beendet das Token⟩ ⟨Sonderzeichen sind separate neue Token⟩ ⟨Sonstige Zeichen setzen das Token fort⟩ ⟨Am Ende des Eingabestrings endet das Token⟩

Manchmal braucht man eine Vorausschau in den String:

⟨Ziffer, Punkt, Leerzeichen beenden ein Token⟩≡

tokenize([S,L|String],at,[[C|Cs]|Rs],Ts) :-
        ziffer(C),punktZeichen(S),leerZeichen(L),
        !,tokenToAtom([S,C|Cs],Ordinalzahl),
        tokenize(String,nt,[Ordinalzahl|Rs],Ts).

Beispiel zur Tokenisierung

⟨Aufrufe und Antworten von tokenize/4 zeigen:⟩≡

trace(tokenize,[call,return]).

Die ASCII-Nummern des Beispielstrings

⟨String als ASCII-Liste⟩≡

10 ?- "Der 2. Tag." =
      [68,101,114,32,50,46,32,84,97,103,46].

sind im Folgenden (von Hand) durch lesbare Zeichen ersetzt:

⟨Aufruf des Tokenizers⟩+≡

[debug]
11 ?- stringToAtoms("Der 2. Tag.",Tokens).
  (7) tokenize([D,e,r, ,2,., ,T,a,g,.],nt,[],_G)
  (8) tokenize([e,r, ,2,., ,T,a,g,.],at,[[D]],_G)
  (9) tokenize([r, ,2,., ,T,a,g,.],at,[[e,D]],_G)
  (10) tokenize([ ,2,., ,T,a,g,.],at,[[r,e,D]],_G)
  (11) tokenize([2,., ,T,a,g,.],nt,[’Der’],_G)
  (12) tokenize([., ,T,a,g,.],at,[[2],’Der’],_G)
  (13) tokenize([T,a,g,.],nt,[’2.’,’Der’],_G)
  (14) tokenize([a,g,.],at,[[T],’2.’,’Der’],_G)
  (15) tokenize([g,.],at,[[a,T],’2.’,’Der’],_G)
  (16) tokenize([.],at,[[g,a,T],’2.’,’Der’],_G)
  (17) tokenize([],nt,[’.’,’Tag’,’2.’,’Der’],_G)

Tokens = [’Der’, ’2.’, ’Tag’, ’.’]

Yes

Lemmatizer

lexAnalyse(Wortliste+)

Die Lemmatisierung oder lexikalische Analyse im engeren Sinn:

(i) Beseitige das Sublexikon einer früheren Lemmatisierung.

(ii) Extrahiere aus einem (Vollform-) Lexikon zu jedem Wort die vorhandene Information (lexikalische Kategorie).

(iii) Speichere diese Information in einem Sublexikon.

Das Lexikon ist eine Faktensammlung in der PROLOG-Datenbank, das Sublexikon wird in die Datenbank geschrieben.

⟨Format der Lexikoneinträge⟩≡

word(Wortform?,Kategorie?).

⟨Format der Sublexikoneinträge⟩≡

lexword(Position?,Wortform?,Kategorie?).

Die Position ist die Position eines Worts in der Eingabefolge.

⟨Lemmatisierer, 1.-3.⟩≡

lexAnalyse(Wortliste) :-
     (gescheitert(lexAnalyse)
     -> retract(gescheitert(lexAnalyse)) ; true),
     retractall(lexword(_,_,_)),
     (ein(satzzeichen) % Option: Satzzeichen erhalten
     -> Ws = Wortliste
     ; entferneSatzzeichen(Wortliste,Ws)),
     lexikalischeAnalyse(Ws,0),
     (ein(filterLexAnalyse) -> (nl,filterLexword); true).

Lemmatisierer (Fors.)

Zähle die Positionen mit und mache entsprechende Einträge in die Datenbank:

⟨Lemmatisierer, 3.⟩≡

lexikalischeAnalyse([W|Ws],N) :-
        eintraege(W,N),
        M is N+1,
        lexikalischeAnalyse(Ws,M).
lexikalischeAnalyse([],_).

⟨Eintrag in lexword/3 machen:⟩≡

eintrag(Wortform,Kategorie,Position) :-
        assertz(lexword(Position,Wortform,Kategorie)).

Um die Einträge zu bilden, macht man in eintraege(W,N):

(i) Falls W eine Zahl (Ordinal-, usw.) ist, gib die Zahlart als Kategorie an.

(ii) Falls W eine Abkürzung ist, sind ihre Kategorien einem Abkürzungslexikon zu entnehmen.

(iii) Falls W ein Satzzeichen ist, dient es als seine Kategorie.

(iv) Falls W Einträge im Vollformenlexikon hat, übertrage diese Kategorien.

(v) Falls W mit Großbuchstabe beginnt, keine Einträge im Vollformenlexikon hat und N der Satzanfang ist, suche Einträge zum W mit kleinem Anfangsbuchstaben.

(vi) Falls zu W keine Information vorhanden ist, gib eine Warnung ’unbekanntes Wort’ aus und erkläre die lexikalische Analyse als gescheitert. (Parser nicht aufrufen!)

⟨Auszüge aus eintraege(W,N):⟩≡

...
eintraege(W,N) :-
        name(W,String),
        abkuerzung(String),
        !,eintraege_abk(String,W,N).
eintraege_abk(String,W,N) :-
        abkuerzung(String,Kategorie),
        eintrag(W,Kategorie,N),
        fail.
eintraege_abk(_String,_W,_N).
...
eintraege(W,N) :-
        word(W,_),
        !,lexKats(W,W,N).
...
lexKats(Orig,W,N) :-
        not(word(Orig,_)),
        write(user,’\nUnbekanntes Wort: ’),write(user,W),
        !,eintrag(W,W,N),
        (gescheitert(lexAnalyse) -> true
          ; assertz(gescheitert(lexAnalyse))).
lexKats(Orig,W,N) :-
        word(Orig,Cat),
        eintrag(W,Cat,N),
        fail.
lexKats(_Orig,_W,_N).

Tagger

Hier kann man

(i) die nach statistischen Kriterien der üblichsten Trigramme (Tripel lexikalischer Kategorien aufeinanderfolgender Wörter) ausgeschlossenen Lesarten beseitigen,

(ii) die nach syntaktischen Kriterien an bestimmten Positionen (Satzanfang, -Ende, vor und nach Satzzeichen) ausgeschlossenen Lesarten beseitigen,

(iii) die nach Meinung des Benutzers ausgeschlossenen Lesarten beseitigen (nützlich zur Fehlersuche).

Bem. Für die Satzgrammatik ist nur die dritte Möglichkeit implementiert (vgl. die Option filterLexAnalyse).

Sprachdefinitionen und Grammatiken Eingabealphabet: Σ := c_1 | c_2 | . . . | c_k. Eingabefolge: w ∈ Σ^∗ :=

⋃_∞ i=0 Σ^n, oft mit Endmarken: $w$ Eingabegrammatik: G = (SynCat, Σ, P, S) mit

(i) endl. Menge SynCat von syntaktischen Kategorien, (ii) endl. Menge Σ von lexikalischen Kategorien (LexCat) (iii) endl. Menge P von grammatischen Regeln (iv) ausgezeichnete Satz- oder Hauptkategorie(n) S ∈ SynCat Wir betrachten statt einer Eingabefolge w von Zeichen gleich w als Ergebnis der Tokenisierung und Lemmatisierung, grob Σ = Alphabet = Terminalsymbole = Token = LexCat.

Entsprechend für die durch G rekursiv definierten Kategorien: X = Rekursionsvariable = Nonterminalsymbole = SynCat. Eine Satzform zu G ist eine Folge α ∈ Cat^∗ von syntaktischen und lexikalischen Kategorien, mit Cat := SynCat ∪ LexCat.

Jede Grammatik G ist eine endliche Definition einer i.a. unendlichen formalen Sprache L(G) ⊆ Σ^∗.

Kontextfreie Grammatiken

P ist eine Menge von Regeln A → α mit A ∈ SynCat und α ∈ Cat^∗.

Jedes (A → α) ∈ P definiert auf Cat^∗ die Relation der Ersetzung eines Vorkommens von A in β durch α: β ⇒_{(A→α)} γ : ⇐⇒ ∃β_1, β_2 ∈ Cat^∗[β ≡ β_1Aβ_2 ∧ γ ≡ β_1αβ_2]

Das wird fortgesetzt zu:

⇒_P := ⋃ { ⇒_{(A→α)}| (A → α) ∈ P } ⇒^∗_P := reflexive transitive Hülle von ⇒_P .

Vereinfachende Schreibweise:

β ⇒^∗ γ oder β →^∗ γ statt β ⇒^∗_P γ.

Damit erklärt man die von G definierte Sprache als L(G) := { w ∈ Σ^∗ | S ⇒^∗_P w }.

Syntaktische Struktur (allgemein)

Eine syntaktische Struktur eines sprachlichen Ausdrucks w ∈ Σ^∗ bezüglich einer Sprachdefinition G = (SynCat, Σ, P, S) ist ein Beweis dafür, daß w ∈ L(G).

Eine grammatische Regel sollte eigentlich eine Konstruktionsregel sein, und jede Konstruktion eines Ausdrucks sollte eine andere syntaktische Struktur sein.

(„Freie“ Konstruktionen, „Äquivalenzklassen von“ Beweisen!)

Syntaktische Struktur (kontextfreie Grammatik) Bei einer kontextfreien Grammatik ist jede Ableitung γ_1 →_{(A1→α1)} γ_2 ⇒_{(A2→α2)} γ_2 ⇒ . . . ⇒_{(An→αn)} γ_n mit S ≡ γ_1 und w ≡ γ_n ein Beweis für w ∈ L(G).

Gewisse dieser Ableitungen gelten als äquivalent (s.u.) Beisp. 1 Die beiden Ableitungen

S ⇒_{(S→AB)} AB ⇒_{(A→a)} aB ⇒_{(B→bc)} abc S ⇒_{(S→AB)} AB ⇒_{(B→bc)} Abc ⇒_{(A→a)} abc sind äquivalent, da sie denselben Syntaxbaum haben: S�

A�

a�

� � � � B�

b� � � � c�

Die beiden Ableitungen

S ⇒_{(S→AB)} AB ⇒_{(A→a)} aB ⇒_{(B→bc)} abc S ⇒_{(S→AC)} AC ⇒_{(C→bc)} Abc ⇒_{(A→a)} abc sind nicht äquivalent für B ≢ C.

Syntaktische Struktur = Syntaxbaum

Jeder Ableitung ordnet man einen Wald zu, d.h. eine Folge angeordneter Bäume mit Knotenmarkierungen in Cat. Für die Folge A_1 · · · A_n ≡ α ∈ (Cat ∪ {ϵ})^∗ setze:

(i) der Wald (der Länge n = |α|) zur Ableitung α ⇒^0 α ist ⟨•A_1, . . . , •A_n⟩.

(ii) der Wald zur Ableitung α ≡ βAγ ⇒_{(A→A1...An)} βA_1 · · · A_nγ ⇒^k δ ist ⟨r_1, . . . , r_{|β|}, s, t_1, . . . , t_{|γ|}⟩, wenn s der Baum A�

s_n� � � � · · ·� � � s_1� � � �

und ⟨r_1, . . . , r_{|β|}, s_1, . . . , s_n, t_1, . . . , t_{|γ|}⟩ der Wald zur Restableitung βA_1 · · · A_nγ ⇒^∗ δ ist.

Zwei Ableitungen heißen äquivalent, wenn ihnen hierdurch derselbe Wald zugeordnet wird.

Ein Syntaxbaum von G ist ein solcher Wald mit Länge 1, d.h. eine Äquivalenzklasse von Ableitungen α ⇒^∗ β mit |α| = 1.

Reguläre Ausdrücke über Σ als Sprachdefinitionen

Die Menge der reguläre Ausdrücke mit Konstanten c aus Σ und Variablen x aus X, kurz reg_Σ(X), ist definiert durch r, s := x | c | 0 | 1 | (r + s) | (r; s) | r^∗ Sind Sprachen A_i ⊆ Σ^∗ für die Variablen x_i in r(x_1, . . . , x_n) gegeben, so erhält man L(r) := L(r)[A_1, . . . , A_n] induktiv durch

L(0) := ∅ L(s + t) := L(s) ∪ L(t) L(1) := {ϵ} L(s ; t) := L(s)L(t) L(c) := {c} L(s^∗) := L(s)^∗ L(x_i) := A_i

Jedes r ∈ reg_Σ (ohne Variablen) können wir als eine kontextfreie Grammatik G_r = (SynCat, Σ, P, S) verstehen, wobei

(i) SynCat sind die Teilausdrücke von r, (ii) P besteht aus folgenden Regeln, für s, t ∈ SynCat:

(s + t) → s s^∗ → 1 (s + t) → t s^∗ → s s^∗ (s; t) → s t 1 → ϵ

(iii) S ist r.

Natürlich erhält man L(r) = L(G_r) für geschlossene reguläre Ausdrücke r, also mit G_r auch vernünftige(!) Syntaxbäume für reguläre Sprachen.

Interpretation durch Wortmengen Die obige Interpretation der regulären Ausdrücke und der grammatischen Kategorien im (unendlichen) Modell

L = (|L|, +^L, 0^L, ·^L, 1^L) = (2^{Σ∗} , ∪, ∅, · , {ϵ}) interpretiert Kategorien A durch Wortmengen L(A) ⊆ Σ^∗, wobei +^L die Vereinigung, ·^L die wortweise Verkettung L(A · B) = L(A) ·^L L(B) := { uv | u ∈ L(A), v ∈ L(B) }, und a ∈ Σ die Einermenge L(a) := a^L := {a} ist.

Interpretation durch Tokenmengen Zur Analyse eines Satzes w braucht man aber nicht alle Ausdrücke einer Kategorie A, sondern nur die in w vorkommenden. Sei w = a_1 · · · a_n, I = {0, 1, . . . , n} und v ∈ Σ^∗. Ein Vorkommen (Token) von v in w ist ein Paar (i, j) ∈ I × I mit v = a_{i+1} · · · a_j . Die regulären Ausdrücke und grammatischen Kategorien kann man relativ zu w im endlichen(!) Modell T = (|T |, +^T , 0^T , ·^T , 1^T ) = (2^{I×I} , ∪, ∅, ∘ , =) durch Tokenmengen T (A) ⊆ I × I interpretieren. Dabei wird a ∈ Σ durch die Menge seiner Vorkommen in w interpretiert, T (a) := a^T := { (i, i + 1) | i ∈ I, a_i = a }, +^T ist die Vereinigung und ·^T das Produkt von Relationen, T (A · B) := T (A) ·^T T (B) := { (i, j) | ∃k ∈ I((i, k) ∈ T (A) ∧ (k, j) ∈ T (B)) } Beisp: In w =_0a_1b_2a_3a_4b_5c_6b_7 ist T (ab) = {(0, 2), (3, 5)}.

EBNF-Grammatiken

Eine EBNF -Grammatik G = (SynCat, Σ, P, S) über Σ ist eine kontextfreie Grammatik mit Regeln

(A → α), wobei α ∈ reg_Σ(SynCat).

Die Menge {A → α_1, . . . , A → α_k} aller A-Regeln von G fassen wir zu einer Gleichung A = α_1 + · · · + α_k zusammen.

Reguläre Gleichungssysteme über Σ

in den Rekursionsvariablen X = {x_1, . . . , x_n} sind Systeme

x_1 = r_1(x_1, . . . , x_n) . . . . x_n = r_n(x_1, . . . , x_n)

mit r_i(x_1, . . . , x_n) ∈ reg_Σ(X). Sie sind eine andere Notation für EBNF-Grammatiken G = (X, Σ, P, x_1) mit passendem P . Definierte Sprache: L_1 für das bzgl. ⊆ kleinste Tupel L = (L_1, . . . , L_n) von Sprachen über Σ mit

L_1 = r_1(L_1, . . . , L_n) . . . . L_n = r_n(L_1, . . . , L_n)

Die nicht-atomaren regulären Teilausdrücke r von r_i bzw. α nennen wir zusammengesetzte Kategorien.

Beispiel Beisp. 2 Die Gleichungssysteme

x = a(x + y)b + 1 y = b(x + 1) + y x = axb + ayb + 1 y = bx + b + y entsprechen der CF-Grammatik mit S := x und den Regeln x → axb, x → ayb, x → ϵ, y → bx, y → b, y → y.

Kleinste Lösung eines Gleichungssystems Sei M das unendlichen Modell L aller Wortmengen über Σ oder das endliche Modell T aller Tokenmengen über w ∈ Σ^∗. Die kleinste Lösung (M_1, . . . , M_n) eines Gleichungssystems x_i = r_i(x_1, . . . , x_n), (1 ≤ i ≤ n) in M erhält man, indem man mit den Mengen (Relationen) M_{i,0} := ∅ (1 ≤ i ≤ n) beginnt und dann den Wert der regulären Ausdrücke ergänzt M_{i,k+1} := r_i(M_{1,k}, . . . , M_{n,k}) ∪ M_{i,k}.

Die Komponenten der kleinsten Lösung (M_1, . . . , M_n) sind dann M_i := ⋃ k∈IN

M_{i,k}, (1 ≤ i ≤ n).

Beachte: in M = T ist die Kette ∅ = M_{i,0} ⊆ M_{i,1} ⊆ . . . ⊆ M_i ⊆ I × I endlich!

NICHT-DETERMINISTISCHE ANALYSE (universell)

Nicht-det. Top-Down-Analyse

Die Top-Down- oder hypothesengesteuerte Analyse geht von der Hypothese, die Eingabe w sei ein Ausdruck einer bestimmten (Start-) Kategorie S, aus und versucht, eine Linksableitung

S ⇒_{lm} α_1 ⇒_{lm} α_2 ⇒_{lm} . . . ⇒_{lm} w

zu finden. D.h. in jedem Schritt α_i ⇒_{lm} α_{i+1} wird die am weitesten links in α_i stehende syntaktische Kategorie

α_i ≡ vBγ ⇒_{(B→β)} vβγ ≡ α_{i+1}

für ein v ∈ Σ^∗, B ∈ SynCat und eine Regel (B → β) ∈ P expandiert:

� � � � � � � S � � � B γ

v � � � � ⇒_{(B→β)}

S

� � � � � � � �γ � �β v � � a) Top-Down-Erkenner nicht-det. Kellerautomat M^G

Konfigurationen: (α, w) ∈ SynCat^∗ × Σ^∗ Anfangskonfiguration: (S, a_1 · · · a_n) Endkonfiguration: (ϵ, ϵ) Befehle:

Expandiere mit (B → β) ∈ P : (Bγ, w) ⊢_{(B→β)} (βγ, w) Vergleiche mit Eingabe a ∈ Σ: (aγ, aw) ⊢_a (γ, w) Programm von M^G: r := (P ∪ Σ)^∗

Prop. 3 w ∈ L(G) ⇐⇒ S ⇒^∗_{lm} w ⇐⇒ (S, w) ⊢_M (ϵ, ϵ)

Beisp. 4 Σ := {a, b, c}, P := {S → AC, A → AB, A → a, B → b, C → BC, C → c}

Linksableitung: Erkennung durch M^G S ⇒_{(S→AC)} AC ⇒_{(A→AB)} ABC ⇒_{(A→a)} aBC ⇒_{(B→b)} abC ⇒_{(C→c)} abc

(S, abc) ⊢_{(S→AC)} (AC, abc) ⊢_{(A→AB)} (ABC, abc) ⊢_{(A→a)} (aBC, abc) ⊢_a (BC, bc) ⊢_{(B→b)} (bC, bc) ⊢_b (C, c) ⊢_{(C→c)} (c, c) ⊢_c (ϵ, ϵ) b) Top-Down-Parser mit Rücksetzungen

Die Grammatikregeln seien numeriert.

Tiefensuche: Ersetze in α das linkeste A ∈ SynCat mit der nächsten möglichen Regel p : (A → α); wenn ein Vergleich mit dem Eingabesymbol scheitert, gehe zurück zur letzten Wahlmöglichkeit und benutze die nächste Regel.

Konfigurationen (α, w, π, p) ∈ Cat^∗ × Σ^∗ × Parse × P^′ Anfangskonfiguration: (S, a_1 · · · a_n, ϵ, 0) Endkonfiguration: (ϵ, ϵ, π, p) Befehle:

Expandiere mit p = (A → α): (Aβ, w, π, p) ⊢_p (αβ, w, pπ, 0) Suche Regel für p ≠ (A → α): (Aβ, w, π, p) ⊢ (Aβ, w, π, p + 1) bzw. für p_{max} = p ≠ (A → α): (Aβ, w, π, p) ⊢ (Aβ, w, π, redo) Lesen der Eingabe a ∈ Σ: (aβ, aw, π, 0) ⊢_a (β, w, aπ, 0) bzw. Lesekonflikt a ≠ b: (aβ, bw, π, 0) ⊢ (aβ, bw, π, redo) bzw. Lesekonflikt ϵ ≠ b: (ϵ, bw, π, 0) ⊢ (ϵ, bw, π, redo) bzw. Lesekonflikt aβ ≠ ϵ: (aβ, ϵ, π, 0) ⊢ (aβ, ϵ, π, redo) Redo mit p = (A → α): (αβ, w, pπ, redo) ⊢ (Aβ, w, π, p + 1) bzw. Stop bei p = p_{max}: (αβ, w, pπ, redo) ⊢ (Aβ, w, π, stop) Redo mit a ∈ Σ: (α, w, aπ, redo) ⊢ (aα, aw, π, redo) Erfolgreiche Analyse: (ϵ, ϵ, π, 0) ⊢ (ϵ, ϵ, π, redo) Nachteil: Divergiert bei Linksrekursion A ⇒^+ Aβ.

Suchbaum der Linksableitungen aus S

Wähle von S ausgehend bei jedem mit A ∈ SynCat markierten ∨-Knoten einen Ast (Regel A → β) und dann bei dem mit β markierten Nachfolger (∧-Knoten) für jeden Nachfolger B wieder einen Ast, usw. ∨ S

∧ α_n � � � . . . � � � ∧ α_1

∨ A_m � � � � � . . . � � � ∨ A_2

∧ β_k � � � . . . � � � ∧ β_1� � � � � � ∨ A_1� � � � � � � �

Eine Analyse wird durch die Folge π der gewählten Regeln und der in der Eingabe gefundenen Blattmarken a ∈ Σ kodiert.

Die Suche nach einer Linksableitung gleicht der Beweissuche in Prolog, wo zuerst die linkesten Teilziele (Konjunktionsglieder) und dann die weiter rechts stehenden bewiesen werden.

⟨Kontextfreie Grammatikregel⟩≡

np(sg,nom) ---> det(sg,nom) ap(sg,nom) n(sg,nom)

⟨Entsprechende Prolog-Klausel⟩≡

np(sg,nom,I,J) :-
   det(sg,nom,I,K), ap(sg,nom,K,L), n(sg,nom,L,J).

mit Differenzlistendarstellung I = [a_{i+1}, . . . , a_j |J] von Teillisten v = a_{i+1}, . . . , a_j der Eingabe.

Beispiel zum Top-Down-Parser

Der Top-Down-Parser durchsucht bei jedem ∨-Knoten zuerst den linkesten Teilbaum.

Beisp. 5 Sei G die Grammatik mit den Regeln

0 : (A → BC), 1 : (B → ab), 2 : (C → c), 3 : (C → AB) Der Parser arbeitet bei Eingabe von abc und Ziel A so:

(A, abc, ϵ, 0) ⊢ (BC, abc, 0, 0) ⊢ (BC, abc, 0, 1) ⊢ (abC, abc, 10, 0) ⊢ (bC, bc, a10, 0) ⊢ (C, c, ba10, 0) ⊢ (C, c, ba10, 1) ⊢ (C, c, ba10, 2) ⊢ (c, c, 2ba10, 0) ⊢ (ϵ, ϵ, c2ba10, 0) (1.Analyse!) ⊢ (ϵ, ϵ, c2ba10, redo) ⊢ (c, c, 2ba10, redo) ⊢ (C, c, ba10, 3) ⊢ (AB, c, 3ba10, 0) ⊢ (BCB, c, 03ba10, 0) ⊢ (BCB, c, 03ba10, 1) ⊢ (abCB, c, 103ba10, 0) ⊢ (abCB, c, 103ba10, redo) ⊢ (BCB, c, 03ba10, 2) ⊢ (BCB, c, 03ba10, 3) ⊢ (BCB, c, 03ba10, redo) ⊢ (AB, c, 3ba10, 1) ⊢ (AB, c, 3ba10, 2) ⊢ (AB, c, 3ba10, 3) ⊢ (AB, c, 3ba10, redo) ⊢ (C, c, ba10, stop)

Implementierung in Prolog ⟨Regelformat⟩≡

(SynCat ---> Cat1, ..., [Word], ..., CatN).

Beisp. 6 ⟨cfg.grm⟩≡

% Startsymbole:
start_symbol(s(pl)).
start_symbol(s(sg)).
% Grammatikregeln:
:- op(1100,xfx,’--->’).
s(sg) ---> a(sg), b(sg).
s(pl) ---> a(sg), b(pl).
b(pl) ---> b(sg), [c], b(pl).
b(pl) ---> b(sg), a(sg).
% Lexikon:
word(a,a(sg)).
word(b,b(sg)).
% word(b(sg),b(sg)). % (f. Kategorie in der Eingabe)

Definierte Sprache:

L(cfg.grm) = {ab} ∪ { a(bc)^nba | n ∈ IN }.

c) Top-Down-Erkenner in Prolog

universell, mit Rücksetzungen

Vor.: G definiert die Prädikate ’--->’/2 und start symbol/1 und ist (unbetrachtet) in den Arbeitsspeicher geladen.

⟨Differenzlistendarstellung von [a_{i+1},...,a_j ] durch I,J:⟩≡

I = [a_i+1,...,a_j|J]

⟨top-down-rec.cfg.pl⟩≡

parse(Atoms) :-
        start_symbol(S),
        parse(S,Atoms,[]).
parse(A,I,J) :-
        (A ---> B), % A ist SynCat mit einer Regel
        tupleToList(B,Bs),
        parse_list(Bs,I,J).
parse(A,I,J) :- % A ist LexCat
        I = [W|J],
        ( word(W,A) % eines Worts im Lexikon
        ; A = [W] ). % oder in einer Regel

⟨tupleToList⟩

parse_list([B|Bs],I,J) :-
        parse(B,I,K), parse_list(Bs,K,J).
parse_list([],I,J) :- I = J.

Hilfsfunktion (auch für andere Parser):

⟨tupleToList⟩≡

tupleToList((B,C),[B|Cs]) :- !, tupleToList(C,Cs).
tupleToList([],[]).
tupleToList([W],[[W]]).
tupleToList(B,[B]).

d) Top-Down-Parser mit explizitem Stapel

Der Erkenner ⟨top-down-rec.cfg.pl ⟩ benutzt die in Prolog eingebaute Beweissuche zur Verwaltung der Alternativen. Merkt man sich auf einem Stapel die benutzten Regeln und gelesenen Token, so kann man daraus eine Analyse rekonstruieren:

⟨top-down-stack.cfg.pl⟩≡

:- op(1100,xfx,’--->’).

parse(Atoms) :-
        start_symbol(S),
        parse(S,[],Parse,Atoms,[]).

parse([W],Stapel,Parse,I,J) :- % Terminal in
        !, I = [W|J], % einer Regel
        Parse = [W|Stapel].
parse((B,Bs),Stapel,Parse,I,J) :- % CFG-Regel
        !, parse(B,Stapel,BStapel,I,K),
        parse(Bs,BStapel,Parse,K,J).

parse(A,Stapel,Parse,I,J) :-
        (A ---> B),
        parse(B,[(A ---> B)|Stapel],Parse,I,J).
parse(A,Stapel,Parse,I,J) :-
        I = [W|J],
        word(W,A),
        Parse = [W|Stapel].

Beispiel

Die Analyse ergibt sich aus dem invertierten Stapel ⟨Beispiel ⟩≡

?- [’cfg.grm’, ’top-down-stack.cfg’].

?- parse(s(A),[],Parse,[a,b,a],[]).

A = pl
Parse = [a, b, (b(pl)--->b(sg), a(sg)),
            a, (s(pl)--->a(sg), b(pl))]

wenn man die Benutzung des Lexikons berücksichtigt:

(s^{pl} , aba) ⊢_spl_{→a}sg_bpl (a^{sg} b^{pl} , aba) ⊢_asg_{→a} (ab^{pl} , aba) ⊢_a (b^{pl} , ba) ⊢_bpl_{→b}sg_asg (b^{sg} a^{sg} , ba) ⊢_bsg_{→b} (ba^{sg} , ba) ⊢_b (a^{sg} , a) ⊢_asg_{→a} (a, a) ⊢_a (ϵ, ϵ)

Einfache Baumanzeige

Wir kodieren Bäume durch Listen in der Form

⟨Baumkodierung: ⟩≡

[Kategorie der Wurzel | Teilbäume]

Eine lesbare Ausgabe der Bäume liefert dann:

⟨printTree.pl⟩≡

printTree(Tree,Indent) :-
    printTree(Tree,5,Indent).

printTree([A,W],I,Indent) :-
    atom(W),!,
    nl, tab(I), writeq(A), tab(2), writeq(W).
printTree([A|Trees],I,Indent) :-
    !, printTree(A,I,Indent),
    J is I + Indent, printTrees(Trees,J,Indent).
printTree(A,I,_) :-
    nl, tab(I), writeq(A).

printTrees([T|Ts],I,Indent) :-
    printTree(T,I,Indent),
    printTrees(Ts,I,Indent).
printTrees([],_I,_Indent).

Hierbei ist I die horizontale Einrückung vom linken Zeilenrand und Indent die relative Einrückung der Teilbäume.

d) Top-Down-Parser (universell, für DCGs)

⟨Einfache und zusammengesetzte Kategorien⟩≡

Cat := Atomic % Einfache Kat.
     | (Cat,Cat) | (Cat;Cat) | {Goal} | [W|Ws] | []

Konstruiert man für alle Cat Baumlisten, so braucht man für die komplexen Kategorien keine Knoten im Syntaxbaum:

⟨top-down.pl⟩≡

:- op(1100,xfx,’--->’).
parse(Atoms,Tree) :-
    start_symbol(S), parse(S,[Tree],Atoms,[]).

parse((B-C),Trees,I,J) :- % (Differenz, s.u.)
    !, parse(B,Trees,I,J), not(parse(C,_,I,J)).
parse((B,C),Trees,I,J) :- % Zusammeng.Kategorien
    !, parse(B,TreesB,I,K), parse(C,TreesC,K,J),
    append(TreesB,TreesC,Trees).
parse((B;C),Trees,I,J) :-
    !, (parse(B,Trees,I,J) ; parse(C,Trees,I,J)).
parse({Goal},Trees,I,J) :-
    !, call(Goal), Trees = [], I = J.
parse([],Trees,I,J) :-
    !, Trees = [], I = J.
parse([W|Ws],Trees,I,J) :-
    !, append([W|Ws],J,I), Trees = [W|Ws].

parse(A,Trees,I,J) :- % Einfache Kategorien
    (A ---> Body),
    parse(Body,SubTrees,I,J), Trees = [[A|SubTrees]].
parse(A,Trees,I,J) :-
    I = [W|J], word(W,A), Trees = [[A,W]].

Beispiel

⟨cfg3.grm⟩≡

start_symbol(’A’).

:- op(1100,xfx,’--->’).

’A’ ---> (((’B’,[c]),’D’) ; (’D’,(’B’;’A’))).
’B’ ---> [a,b].
’C’ ---> ’A’,’B’.
’D’ ---> [d].

word(c,’C’).

Anwendung

⟨Beispiel ⟩+≡

?- [’top-down’,’cfg3.grm’,printTree].
?- parse([d,a,b,c,d],Baum), printTree(Baum,4).

    ’A’
       ’D’
          d
       ’A’
          ’B’
             a
             b
          c
          ’D’
             d

Baum = [’A’, [’D’, d], [’A’, [’B’, a, b], c, [’D’, d]]]

Beispiel: Definite Clause Grammar Beisp. 7 Grammatik cfg.grm kompakter als DCG formuliert: ⟨dcg.grm⟩≡

% Startsymbole:
start_symbol(s(pl)).
start_symbol(s(sg)).
% Grammatikregeln:
:- op(1100,xfx,’--->’).
s(Num) ---> a(sg),
        ( b(sg), {Num = sg}
        ; b(pl), {Num = pl}
        ).
b(pl) ---> b(sg), ([c], b(pl) ; a(sg)).
% Lexikon:
word(a,a(sg)).
word(b,b(sg)).

Definierte Sprache:

L(dcg.grm) = L(cfg.grm) = {ab} ∪ { a(bc)^nba | n ∈ IN }. Beachte, wie mit {Goal} das Merkmal Num abhängig vom benutzten Disjunktionsglied belegt und vererbt wird.

⟨Beispielanalyse⟩≡

time(parse([a,b,c,b,a],Baum)), printTree(Baum,4).

    s(pl)
       a(sg)
          a
       b(pl)
          b(sg)
             b
          c
          b(pl)
             b(sg)
                b
             a(sg)
                a

% 150 inferences, 0.00 CPU in 0.00 seconds
  (0% CPU, Infinite Lips)

Baum = [s(pl), [a(sg), a],
        [b(pl), [b(sg), b], c, [b(pl), [...|...]|...]]] ;

Beispiel für Differenzkategorien

Ein Top-Down-Parser kann mit (C-D) ausschließen, daß eine Konstituente C eine bestimmte Form D hat: z.B. darf ein Genitivattribut keine pronominalisierte Nominalphrase sein.

⟨cfg4.grm⟩≡

start_symbol(’NP’).
:- op(1200,xfx,’--->’).

’NP’ ---> ’proNP’ ; ’Det’, ’N’, ([] ; ’NPgen’).
’NPgen’ ---> ’NP’ - ’proNP’. % Differenzkategorie!
’Det’ ---> ’Art’ ; poss.
’Art’ ---> [die] ; [der] ; [das] ; [des].
’N’ ---> [tür] ; [haus] ; [hauses].
’proNP’ ---> [er].
’poss’ ---> [seine] ; [seines].

⟨Nur nicht-pronominalisierte NPs sind Genitivattribute⟩≡

?- parse(’NP’,Baum,[er],[]).
Baum = [[’NP’, [proNP, er]]]

?- parse(’NP’,Baum,[seines,hauses],[]).
Baum = [[’NP’, [’Det’, [poss, seines]], [’N’, hauses]]]

?- parse(’NP’,Baum,[die,tür,seines,hauses],[]).
Baum = [[’NP’, [’Det’, [’Art’, die]], [’N’, tür],
               [’NPgen’, [’NP’, [...|...]|...]]]]

?- parse(’NP’,Baum,[die,tür,er],[]).
No

Schwieriger: Tests an die Struktur erkannnter Konstituenten!

Nicht-det. Bottom-Up-Analyse

Die Bottom-Up- oder Shift-Reduce- oder datengesteuerte Analyse versucht, eine Rechtsableitung S ⇒_{rm} α_1 ⇒_{rm} α_2 ⇒_{rm} . . . ⇒_{rm} w zu finden. D.h. in jedem Schritt α_i ⇒_{rm} α_{i+1} benutzt man die am weitesten rechts in α_i stehende syntaktische Kategorie α_i ≡ αBv ⇒_{(B→β)} αβv ≡ α_{i+1} für ein v ∈ Σ^∗, B ∈ SynCat und eine Regel (B → β) ∈ P .

Die Ableitungsschritte werden in umgekehrter Reihenfolge durch Reduktionen von α_{i+1} zu α_i rekonstruiert:

α_{i+1}

� � � � α β v

⇒^{−1}_{(B→β)}

� � α_i

� � � � α B � � v

In α_{i+1} ≡ αβv muß man den Henkel β finden und reduzieren.

a) Bottom-Up-Erkenner (nicht-det. Kellermaschine M_G)

Konfigurationen: (α, w) ∈ Cat^∗ × Σ^∗ Anfangskonfiguration: (ϵ, a_1 · · · a_n) Endkonfiguration: (S, ϵ) Befehle:

Reduziere mit (B → β) ∈ P : (αβ, w) ⊢_{(B→β)} (αB, w) Lese a ∈ Σ von der Eingabe: (α, aw) ⊢_a (α a, w) Programm von M_G: r := (P ∪ Σ)^∗

Prop. 8 w ∈ L(G) ⇐⇒ S ⇒^∗_{rm} w ⇐⇒ (ϵ, w) ⊢_{MG} (S, ϵ)

Beisp. 9 Σ := {a, b, c}, P := {S → AC, A → AB, A → a, B → b, C → BC, C → c} Rechtsableitung: Erkennung durch M_G S ⇒_{(S→AC)} AC ⇒_{(C→c)} Ac ⇒_{(A→AB)} ABc ⇒_{(B→b)} Abc ⇒_{(A→a)} abc

(ϵ, abc) ⊢_a (a, bc) ⊢_{(A→a)} (A, bc) ⊢_b (Ab, c) ⊢_{(B→b)} (AB, c) ⊢_{(A→AB)} (A, c) ⊢_c (Ac, ϵ) ⊢_{(C→c)} (AC, ϵ) ⊢_{(S→AC)} (S, ϵ) b) Bottom-Up-Parser (deterministisch, mit Rücksetzungen)

Prinzip: Falls möglich, reduziere den Stapel mit der gegenwärtigen (numerierten) Grammatikregel. Nach jeder Reduktion versuche wieder zu reduzieren. Falls keine Reduktion möglich ist, schiebe das nächste Eingabesymbol auf den Stapel, und versuche wieder zu reduzieren. Falls kein Eingabesymbol mehr vorhanden ist, setze bis zur letzten Alternative zurück und versuche es mit der nächsten Regel. Abstrakte Maschine:

Konfigurationen (α, w, π, p) ∈ Cat^∗ × Σ^∗ × Parse × P^′ Anfangskonfiguration: (ϵ, a_1 · · · a_n, ϵ, 0) Befehle: (p = (B → β)) Reduziere γ ≡ αβ: (γ, w, π, p) ⊢_p (αB, w, pπ, 0) Nächste Regel, γ ≢ αβ: (γ, w, π, p) ⊢ (γ, w, π, p + 1) bzw. Shift: (γ, aw, π, p_{max}) ⊢_a (γa, w, π, 0) bzw. Lesekonflikt: (γ, ϵ, π, p_{max}) ⊢ (γ, ϵ, π, redo) Redo mit p ≠ p_{max}: (αB, w, pπ, redo) ⊢ (αβ, w, π, p + 1) Redo mit p = p_{max}: (αB, w, pπ, redo) ⊢ (αβ, w, π, redo) Redo mit a ∈ Σ: (αa, w, π, redo) ⊢ (α, aw, π, redo) Erfolgreiche Analyse: (S, ϵ, π, 0) ⊢ (S, ϵ, π, redo) Stop: (ϵ, w, ϵ, redo) ⊢ (ϵ, w, ϵ, stop)

Nachteil: Divergiert bei A ⇒^+ A und B → ϵ Exponentieller Aufwand in |w|

Beispiel zum Bottom-Up-Parser

Beisp. 10 Sei G die Grammatik mit den Regeln

0 : (A → BC), 1 : (B → ab), 2 : (C → c), 3 : (C → AB) Der Parser arbeitet bei Eingabe von abc so:

(ϵ, abc, ϵ, 0) ⊢ (ϵ, abc, ϵ, 1) ⊢ (ϵ, abc, ϵ, 2) ⊢ (ϵ, abc, ϵ, 3) ⊢ (a, bc, ϵ, 0) ⊢ (a, bc, ϵ, 1) ⊢ (a, bc, ϵ, 2) ⊢ (a, bc, ϵ, 3) ⊢ (ab, c, ϵ, 0) ⊢ (ab, c, ϵ, 1) ⊢ (B, c, 1, 0) ⊢ (B, c, 1, 1) ⊢ (B, c, 1, 2) ⊢ (B, c, 1, 3) ⊢ (Bc, ϵ, 1, 0) ⊢ (Bc, ϵ, 1, 1) ⊢ (Bc, ϵ, 1, 2) ⊢ (Bc, ϵ, 1, 2) ⊢ (BC, ϵ, 21, 0) ⊢ (A, ϵ, 021, 0) (1.Analyse) ⊢ (A, ϵ, 021, 1) ⊢ (A, ϵ, 021, 2) ⊢ (A, ϵ, 021, 3) ⊢ (A, ϵ, 021, redo) ⊢ (BC, ϵ, 21, 1) ⊢ (BC, ϵ, 21, 2)

Beispiel zum Bottom-Up-Parser . . . und weitersuchen

⊢ (BC, ϵ, 21, 3) ⊢ (BC, ϵ, 21, redo) ⊢ (Bc, ϵ, 1, 3) ⊢ (Bc, ϵ, 1, redo) ⊢ (abc, ϵ, ϵ, 2) ⊢ (abC, ϵ, 2, 0) ⊢ (abC, ϵ, 2, 1) ⊢ (abC, ϵ, 2, 2) ⊢ (abC, ϵ, 2, 3) ⊢ (abC, ϵ, 2, redo) ⊢ (abc, ϵ, ϵ, 3) ⊢ (abc, ϵ, ϵ, redo) ⊢ (ab, c, ϵ, redo) ⊢ (a, bc, ϵ, redo) ⊢ (ϵ, abc, ϵ, redo) ⊢ (ϵ, abc, ϵ, stop)

Nachteile

Divergenz bei Zyklen A ⇒^+ A Divergenz bei Löschregeln A ⇒ ϵ

Der Parser divergiert z.B., wenn es eine Löschregel p : (A → ϵ), aber keine andere Löschregel und keine Regel B → βA gibt:

(γ, w, π, p) ⊢ (γA, w, pπ, 0) ⊢ (γA, w, pπ, 1) ⊢ (γA, w, pπ, 2) ⊢ . . . ⊢ (γA, w, pπ, p) ⊢ (γAA, w, ppπ, 0) ⊢ . . .

c) Bottom-Up-Parser in Prolog (mit Rücksetzungen, für DCG-Regeln)

Ein Bottom-Up-Parser für DCGs spezialisiert beim Reduzieren die rechte Seite (zusammengesetzte Kategorie) der Regel zu einer Folge einfacher Kategorien und entfernt diese vom Stapel:

⟨bottom-up.pl⟩≡

parse(Ws,Tree) :-
    parse(Ws,[],[Tree]),
    Tree = [S|_], start_symbol(S).

parse([W|Ws],Stack,Reduced) :-
    ( reduce(Stack,RedStack),
        parse([W|Ws],RedStack,Reduced)
    ; shift(W,Stack,WStack),
        parse(Ws,WStack,Reduced) ).
parse([],Stack,Reduced) :-
    ( reduce(Stack,RedStack),
        parse([],RedStack,Reduced)
    ; Reduced = Stack ).

shift(W,Stack,[Tree|Stack]) :-
    ( word(W,A), Tree = [A,W]
    ; A = [W], Tree = [A] ).
reduce(Stack,Reduced) :-
    (B ---> Beta),
    roots(Beta,Roots),
    split(Stack,Roots,BetaForest,AlphaForest),
    reverse(BetaForest,BetaTrees),
    Tree = [B|BetaTrees],
    Reduced = [Tree|AlphaForest].

⟨roots.pl⟩

Die Spezialisierung der rechten Regelseite erfolgt durch eine Verallgemeinerung von tupleToList/2:

⟨roots.pl⟩≡

roots((A,B),Cats) :-
    !, roots(A,CatsA), roots(B,CatsB),
    append(CatsA,CatsB,Cats).
roots((A;B),Cats) :-
    !, (roots(A,Cats) ; roots(B,Cats)).
roots([W|Ws],Cats) :-
    !, roots(Ws,CatsWs), Cats = [[W]|CatsWs].
roots([],Cats) :-
    !, Cats = [].
roots({Code},Cats) :-
    !, call(Code), Cats = [].
roots(Cat,[Cat]).

Beim Aufspalten des Forest belegt append die unbekannten Teilbäume in BetaForest:

⟨bottom-up.pl⟩+≡

split(Forest,Beta,BetaForest,AlphaForest) :-
    reverse(Beta,BetaRev),
    newTrees(BetaRev,BetaForest),
    append(BetaForest,AlphaForest,Forest).

newTrees([A|As],[[A|_]|Ts]) :-
    newTrees(As,Ts).
newTrees([],[]).

⟨Bem. unvollständige Behandlung von Differenzkategorien⟩≡

roots((A-B),Cats) :-
    !, roots(A,Cats), not(roots(B,Cats)).

Das verbietet leider nicht, daß die Eingabe eine A-Phrase ist, aber auch eine B-Phrase mit anderen direkten Konstituenten.

Beispiel: Bottom-Up-Parsen

⟨cfg8.grm⟩≡

start_symbol(’A’).
:- op(1100,xfx,’--->’).
’A’ ---> ’B’, ’C’.
’B’ ---> [a], ’B’.
’C’---> ’A’, ’B’.
word(b,’B’).
word(c,’C’).

⟨Analyse: ⟩≡

?- [’cfg8.grm’, ’sr.nd.pl’],
   trace([shift,reduce],[exit]).

?- parse([a,b,c],Tree), printTree(Tree,8).
Exit (10) shift(a, [], [[[a]]])
Exit (11) shift(b, [[[a]]], [[’B’, b], [[a]]])
Exit (12) reduce([[’B’, b], [[a]]],
                  [[’B’, [[a]], [’B’, b]]])
Exit (13) shift(c, [[’B’, [[a]], [’B’, b]]],
                 [[’C’, c], [’B’, [[a]], [’B’, b]]])
Exit (14) reduce([[’C’, c], [’B’, [[a]], [’B’, b]]],
                  [[’A’, [’B’, [[a]], [’B’, b]],
                         [’C’, c]]])
        ’A’
           ’B’
              a
              ’B’
                 b
           ’C’
              c

Tree = [’A’, [’B’, [[a]], [’B’, b]], [’C’, c]]

Aufwand der Bottom-Up-Analyse mit Rücksetzungen

Es gibt kontextfreie Grammatiken G, bei denen jedes w ∈ L(G) nur eine Analyse hat, der Aufwand zu ihrer Berechnung aber O(2^{|w|}) Schritte erfordert.

Beisp. 11 G habe die Regeln 0:(A ---> a) , 1:(A --->

aB), 2:(B ---> aA) . Für n = 1, 2, 3, . . . hat a^n die Analyse

A�

a�

B�

a� � � � A�

a�

A�

a� � � � B�

a� � � � A�

a�

· · ·

Braucht die deterministische Bottom-Up-Analyse f(2n + 1) Schritte zur Analyse von a^{2n+1}, so braucht die von a^{2n+2} mindestens 3 · f(2n + 1) Schritte:

(ϵ, a^{2n+1}a) ⊢^{f(2n+1)} (A, a) ⊢^{≥f(2n+1)} (a^{2n+1}, a) (redo) ⊢ (aa^{2n+1}, ϵ) ⊢^{≥f(2n+1)} (aA, ϵ)

⊢ (B, ϵ).

Daher ist f(2n + 2) ≃ 3 · f(2n + 1) und analog f(2n + 1) ≃ 3 · f(2n), also f(n) ≃ 3^n.

c’) Bottom-Up-Parser in Prolog

(mit Speichern der Wälder)

In forests(I,Roots,Forest) werden die Baumfolgen Forest mit Wurzelmarken Roots gesammelt, die den Teil a_1 · · · a_i der Eingabe a_1 · · · a_n abdecken.

Wälder und ihre Wurzelfolgen werden in invertierter Reihenfolge angegeben, damit man mit Listenoperationen auf die Enden zugreifen kann.

⟨bottom-up.forests.pl⟩≡

% Voraussetzung: CFG-Regeln, Lexikon abgetrennt.
:- dynamic forests/3.

parse(Ws, Tree) :-
        retractall(forests(_,_,_)),
        add(forests(0,[],[])),
        reduce(0,Ws),
        start_symbol(S),
        length(Ws,N),
        forests(N,[S],[Tree]).

shift(I,[W|Ws]) :-
        word(W,A),
        forests(I,Roots,Forest),
        J is I + 1,
        add(forests(J,[A|Roots],[[A,W]|Forest])),
        fail.
shift(I,[W|Ws]) :-
        J is I + 1, reduce(J,Ws).
shift(I,[]).

Bottom-Up-Parser (Forts.)

(mit Speichern der Wälder)

⟨bottom-up.forests.pl⟩+≡

reduce(I,_Ws) :-
    forests(I,Roots,Forest),
    reduce(I,Roots,Forest),
    fail.
reduce(I,Ws) :-
    shift(I,Ws).

reduce(I,Roots,Forest) :-
    (B ---> Beta),
    roots(Beta,Bs),
    reverse(Bs,BetaRev),
    append(BetaRev,AlphaRev,Roots),
    length(BetaRev,N),
    length(BetaForest,N), % N unbekannte Baeume
    append(BetaForest,AlphaForest,Forest),
    reverse(BetaForest,BetaTrees),
    NewForest = [[B|BetaTrees]|AlphaForest],
    NewRoots = [B|AlphaRev],
    add(forests(I,NewRoots,NewForest)),
    reduce(I,NewRoots,NewForest),
    fail.
reduce(I,_Roots,_Forest).

add(T) :- (known_instance(T) -> true ; assertz(T)).
known_instance(Term) :-
    \+ \+ (numbervars(Term,const,0,_), Term).

⟨roots.pl⟩

Vergleich des Aufwands am Beispiel

⟨cfg9.grm⟩≡

start_symbol(’A’).
:- op(1100,xfx,’--->’).
’A’ ---> a, ’B’.
’B’ ---> a, ’A’.
’A’ ---> a.
word(a,a).

Das Speichern der Wälder erspart es zwar, die Bäume abzubauen, aber mit Abspeichern und Suchen dauert es länger:

⟨Aufwand mit sr.pl, ohne explain-Teile⟩≡

?- time(parse([a,a,a,a,a,a,a,a,a],Baum)).
% 277,648 inferences, 2.56 CPU in 2.56 seconds
  (100% CPU, 108456 Lips)

⟨Aufwand mit sr.nd.pl⟩≡

?- time(parse([a,a,a,a,a,a,a,a,a],Baum)).
% 1,702,425 inferences, 2.01 CPU in 2.01 seconds
  (100% CPU, 846978 Lips)

Zwar hat jede Eingabe nur eine Analyse, aber es gibt viele unvollständige Analysen, die auch gespeichert werden.

Bem. Man kann Bäume kompakter darstellen, indem man Bäume mit gleichen Blättern und gleicher Wurzel an den Blättern und der Wurzel identifiziert. Analog für Wälder.

Nichtdeterministische Left-Corner-Analyse

Die (bottom-up) Left-Corner-Analyse mit der (top-down) Erwartung S erweitert partielle Analysen auf zwei unterschiedliche Weisen:

(i) Falls ein Teilwort a_{i+1} · · · a_j der Eingabe durch

� � � � � � � � • B

als Ausdruck der Kategorie B analysiert und B die linke Ecke einer Grammatikregel (A → Bβ) ∈ P ist, so wird die Teilanalyse mit dieser Regel fortgesetzt zu A�

� � � � � � � � • B� � � � β�

und die Kategorien von β bilden neue Erwartungen für die Resteingabe ab a_{j+1}.

(ii) Falls B die nächste Erwartung einer partiellen Analyse A�

β� � � � � B� � � �

� � � � � � � �

� � � · · ·� � �

� � � � � � � � � � � �

von a_{i+1} · · · a_k ist, wird sie um eine vollständige Analyse

� � � � � � � � • B

von a_{k+1} · · · a_j der Kategorie A zu einer partiellen Analyse von a_{i+1} · · · a_j fortgesetzt: A�

β�

� � � � �

� � � � � � � � • B � � � � �

� � � � � � � � � � � · · ·� � � � �

� � � � � � � � � � � � �

(iii) Durch Lesen des nächsten Eingabesymbols a erhält man eine vollständige Analyse, • a, eines Teilworts.

a) Left-Corner-Erkenner (nicht-det. Kellermaschine)

Eine gepunktete Regel ist Grammatikregel mit einem Trennpunkt auf der rechten Regelseite. Die Menge Ṗ := { (A → α · β) | (A → αβ) ∈ P }

enthält die gepunkteten Regeln zu den Regeln von P . Prinzip: (i) Lies ein Eingabesymbol a, (ii) reduziere ϵ, (iii) erweitere die vollständige Analyse eines B-Ausdrucks um eine Regel (A → Bβ) mit linker Ecke B, oder (iv) erweitere eine partielle Analyse mit Erwartung B für eine Position um die vollständige Analyse eines B-Ausdrucks an dieser Position. Konfigurationen: (γ, w) ∈ (Cat ∪ Ṗ )^∗ × Σ^∗ Anfangskonfiguration: (ϵ, a_1 · · · a_n) Endkonfiguration: (S, ϵ) Befehle:

Eingabe lesen: (γ, aw) ⊢_a (γa, w) ϵ reduzieren: (γ, w) ⊢_{(A→ϵ)} (γA, w) Partiell reduzieren: (γB, w) ⊢_{(A→Bβ)} (γ(A → B · β), w) Vervollständigen: (γ(A → α · Bβ)B, w) ⊢ (γ(A → αB · β), w) Vervollständigen: (γ(A → α·), w) ⊢ (γA, w)

(A → α·β) oben auf dem Stapel bedeutet: beim Versuch, einen A-Ausdruck zu erkennen, wurden Ausdrücke der Kategorien α erkannt und Ausdrücke der Kategorien β werden noch gesucht.

Beispiel zur Left-Corner-Erkennung

Für die Grammatik mit den Regeln (S → AC), (A → AB), (A → a), (B → bc), (C → ϵ) und die Eingabe abc ist dies eine akzeptierende Berechnung: (ϵ, abc) ⊢_a (a, bc)

⊢_{(A→a)} ((A → a·), bc) ⊢ (A, bc) ⊢_{(A→AB)} ((A → A · B), bc) ⊢_b ((A → A · B)b, c) ⊢_{(B→bc)} ((A → A · B)(B → b · c), c) ⊢ ((A → A · B)(B → b · c)c, ϵ) ⊢ ((A → A · B)(B → bc·), ϵ) ⊢ ((A → A · B)B, ϵ) ⊢ ((A → AB·), ϵ) ⊢ (A, ϵ) ⊢_{(S→AC)} ((S → A · C), ϵ) ⊢_{(C→ϵ)} ((S → A · C)C, ϵ) ⊢ ((S → AC·), ϵ) ⊢ (S, ϵ)

Aufbau des Syntaxbaums (Left-Corner-Analyse)

a� A� a�

A� A� a� � � � � � B� A� A� a� � � � � � B� b�

A� A� a� � � � � � B� B� b� � � � � c�

A� A� a� � � � � � B� B� b� � � � � c� c�

A� A� a� � � � � � B� b� � � � � c�

S� A� A� a� � � � � � B� b� � � � � c�

� � � � � C� S� A� A� a� � � � � � B� b� � � � � c�

� � � � � C� C� ϵ�

S� A� A� a� � � � � � B� b� � � � � c�

� � � � � C� ϵ� c) Left-Corner-Parser (mit Rücksetzungen)

Aufg. 1 Definiere aus dem nicht-deterministischen Left-Corner- Erkenner einen deterministischen Left-Corner-Parser (mit Rücksetzungen).

Frage 1 Terminiert der Left-Corner-Parser (mit Rücksetzungen) bei jeder kontextfreien Grammatik?

c) Leftcorner-Parser in Prolog (für DCGs, mit Rücksetzungen)

Der Parser benutzt wie der Erkenner einen Stapel mit gepunkteten Regeln (A → α · β), aber merkt sich statt der erkannten Kategorien α die (invertierte) Liste entsprechender Bäume.

⟨leftcorner.pl⟩≡

parse(Words,Tree) :-
        parse(Words,[],[Tree]),
        Tree = [S|_], start_symbol(S).

parse([W|Words],Stack,Reduced) :-
        ( ( complete(Stack,RedStack)
          ; predict(Stack,RedStack)
          ), parse([W|Words],RedStack,Reduced)
        ; shift(W,Stack,RedStack),
            parse(Words,RedStack,Reduced)
        ).
parse([],Stack,Reduced) :-
        ( ( complete(Stack,RedStack)
          ; predict(Stack,RedStack)
          ), parse([],RedStack,Reduced)
        ; Reduced = Stack ).

shift(W,Stack,[Tree|Stack]) :-
        ( word(W,A), Tree = [A,W]
        ; A = [W], Tree = [A] ).

Left-Corner-Parser in Prolog (Forts.)

Eine partielle Analyse mit Erwartungskategorie kann durch einen passenden Ausdruck vervollständigt werden:

⟨leftcorner.pl⟩+≡

complete([[B|Trees],(C ---> RevTrees,[B|Beta])|Stack],
         [(C ---> [[B|Trees]|RevTrees],Beta)|Stack]).

Eine partielle Analyse ohne Erwartungskategorie kann in eine vollständige Analyse umgewandelt werden:

⟨leftcorner.pl⟩+≡

complete([(C ---> RevTrees,[])|Stack],
         [[C|Trees]|Stack]) :-
        reverse(RevTrees,Trees).

Das leere Wort am Anfang der Resteingabe kann zu einem Ausdruck vervollständigt werden:

⟨leftcorner.pl⟩+≡

complete(Stack,[[C]|Stack]) :-
        (C ---> []).

Ein gefundener Ausdruck kann als linke Ecke einer partiellen Analyse der Resteingabe dienen:

⟨leftcorner.pl⟩+≡

predict([[B|Trees]|Stack],Reduced) :-
        (A ---> Alpha),
        roots(Alpha,[B|Beta]),
        Reduced = [(A ---> [[B|Trees]],Beta)|Stack].

⟨roots.pl⟩ DCG-Regeln ersetzt roots/2 quasi in mehrere CFG-Regeln.

Beispiel zum Left-Corner-Parser

⟨cfg.lc.grm⟩≡

start_symbol(s).

% Grammatikregeln:
:- op(1100,xfx,’--->’).
s ---> ap,cp.
ap ---> ap,bp.

% Lexikon:
word(a,ap). word(b,bp). word(c,cp).

⟨Trace⟩≡

?- trace(parse(_,_,_),[all]).
?- parse([a,b,c],Tree).
 (8) parse([a,b,c], [], [Tree])
 (9) parse([b,c], [[ap,a]], [Tree])
 (10) parse([b,c], [(s--->[[ap,a]],[cp])], [Tree])
 ... Fail: (10) parse([b,c], [(s---> ..)], [Tree])
 (10) parse([b,c],[(ap--->[[ap,a]],[bp])], [Tree])
 (11) parse([c],[[bp,b], (ap--->[..],[bp])],[Tree])
 (12) parse([c],[(ap--->[[bp,b],[ap,a]],[])],[Tree])
 (13) parse([c],[[ap,[ap,a],[bp,b]]], [Tree])
 (14) parse([c],[(s--->[[ap,..]],[cp])],[Tree])
 (15) parse([],[[cp,c],(s--->[[ap,..]],[cp])],[Tree])
 (16) parse([],[(s--->[[cp,c],[ap,..]],[])],[Tree])
 (17) parse([],[[s,[ap,[ap,a],[bp,b]],[cp,c]]],[Tree])
 (17) parse([],[[s,[ap,[ap,a],[bp,b]],[cp,c]]],
               [[s,[ap,[ap,a],[bp,b]],[cp,c]]])
 (16) .. (8) parse([a,b,c],[],
                   [[s,[ap,[ap,a],[bp,b]],[cp,c]]])

Tree = [s, [ap, [ap,a], [bp,b]], [cp,c]] ;

d) Leftcorner-Erkenner in Prolog (mit Rücksetzungen)

Manchmal schreibt man den Parser mit Prädikaten parse/3 und leftcorner/4 so, daß bei der Eingabe a_1 · · · a_n gilt:

parse(A, i, j) ⇐⇒ A ⇒^∗ a_{i+1} · · · a_j leftcorner(B, A, i, j) ⇐⇒ A ⇒^∗ Ba_{i+1} · · · a_j

⟨leftcorner.shieber.pl⟩≡

parse(Words) :-
        start_symbol(S), parse(S,Words,[]).
parse(A,I,J) :-
        scan(B,I,K),
        leftcorner(B,A,K,J).

scan(B,I,K) :-
        ( (B ---> []), I=K
        ; (word(W,B) ; B=[W]), I = [W|K] ).
leftcorner(B,A,I,J) :-
        ( B=A, I=J
        ; (C ---> Gamma),
            roots(Gamma,[B|Beta])
            parse_list(Beta,I,K),
            leftcorner(C,A,K,J)
        ).
parse_list([B|Bs],I,J) :-
        parse(B,I,K), parse_list(Bs,K,J).
parse_list([],I,J) :- I = J.

⟨roots.pl⟩

Nichtdeterministische Head-Corner-Analyse

In sog. „X̄-Regeln“ (XP ---> ... X ...) soll die Hauptkonstituente X die Kategorien (ggf.: und Anordnungen) der Nebenkonstituenten ... bestimmen, z.B. das Verb seine Objekte. In jeder Grammatikregel sei eine Hauptkonstituente bestimmt. ⟨Regelformat⟩≡

(SynCat ---> Cat1,...,Cati,...,Catn).

Wir definieren die Hauptkonstituente B durch ein Prädikat: ⟨Hauptrelation⟩≡

head((SynCat ---> Alpha,C,Beta),Alpha,B,Beta).

Beim Head-Corner-Parser soll eine erkannte Phrase nicht als linke Ecke, sondern als Kopfkonstituente verwendet werden.

Vorteil

(i) Linguistisch attraktiv: Kopf-Komplement-Schema (ii) Während die Kategorie der linken Ecke von der Anordnung abhängt, ist die Position des Kopfs (meist) fest, die Kopfrelation also unabhängig von Argumentstellungen berechenbar. (Daher auch für Metaregeln geeignet.)

Nachteil

(i) Kompliziert zu implementieren: Parserichtung rechts-links im Linkskontext, links-rechts im Rechtskontext des Kopfs (ii) Praktische Effizienz?

a) Head-Corner-Erkenner in Prolog

Man definiert induktiv die Relation

headcorner(B, A, i, i^′, j^′, j ) : ⇐⇒ A ⇒^∗ a_{i+1} · · · a_i′ Ba_j′_{+1} · · · a_j .

⟨headcorner-rec.pl⟩≡

parse(A,I,J) :-
    (head(B,A) ; B=A),
    leaf(B,I,I2,J2,J),
    headcorner(B,A,I,I2,J2,J).

headcorner(B,A,I,I2,J2,J) :- % A
    ( I=I2, J2=J, B=A % / \
    ; (C ---> Gamma), % / \
      head((C ---> Gamma),Alpha,B,Beta),% i--- C ---j
      reverse(Alpha,AlphaRev), % / \
      parse_list_rev(AlphaRev,I,I3,I2), % / \
      parse_list(Beta,J2,J3), % / \
      headcorner(C,A,I,I3,J3,J) % i3-i2 B j2-j3
    ).

Mit leaf/5 sucht man im Teilwort a_{i+1} · · · a_j nach möglichen lexikalischen Köpfen a_i′_{+1} = a_j′ :

⟨headcorner-rec.pl⟩+≡

leaf(A,I,I2,J2,J) :-
     ( (A ---> []),
         I=I2, I2=J2, J2=J
     ; append(_,I2,I), I2 = [W|J2], append(_,J,J2),
        (word(W,A) ; A = [W])
     ).

Die Grammatik bestimmt die Hauptkonstituente(n) einer Regel, hier grob durch head/2 allein aus der Phrasenkategorie:

⟨headcorner-rec.pl⟩+≡

head((C ---> Gamma),Alpha,B,Beta) :-
      roots(Gamma,GammaList),
      head(B,C),
      append(Alpha,[B|Beta],GammaList).

In parse list rev(AlphaRev,+I,-K,+J) sucht man K zwischen I und J mit parse(Alpha,K,J):

⟨headcorner-rec.pl⟩+≡

parse_list_rev([],_,J,J).
parse_list_rev([A|As],I,K,J) :-
        parse_rev(A,I,L,J),
        parse_list_rev(As,I,K,L).

parse_rev(A,I,K,J) :-
        append(_,K,I), append(_,J,K),
        parse(A,K,J).

parse_list([B|Bs],I,J) :-
        parse(B,I,K), parse_list(Bs,K,J).
parse_list([],I,J) :- I = J.

⟨roots.pl⟩

Aufg. 2 Erweitere diesen Erkenner zu einem Parser, d.h. ergänze die Ausgabe eines Syntaxbaums.

Beispiel zur Head-Corner-Erkennung

⟨cfg.hc.grm⟩≡

start_symbol(s).
start_symbol(nom).

% Grammatikregeln:
:- op(1100,xfx,’--->’).
s ---> nom, v, akk, rs.
s ---> nom, v, akk.
rs ---> rnom, akk, v.
rs ---> rakk, nom, v.

nom ---> det(nom), n(nom)).
akk ---> det(akk), n(akk)).

% Hauptrelation:
head(v,s). head(v,rs).
head(n(nom),nom). head(n(akk),akk).

% Lexikon:
word(er,nom). word(ihn,akk).
word(der,det(nom)). word(den,det(akk)).
word(’Mann’,n(nom)). word(’Mann’,n(akk)).
word(der,rnom). word(den,rakk).
word(kennt,v). word(grüßt,v).

Bem. Zur Beschleunigung sollte man die transitive Hülle von

{(B,C)| head((C ---> ), ,B, )} vorberechnen und in parse/3

ausnutzen.

Beispiel zur Head-Corner-Erkennung (Forts.)

⟨Tracer einschalten⟩≡

100 ?- trace([headcorner,parse],[+call,+exit]).

⟨Eingabesatz ⟩≡

101 ?- parse(s,[der,’Mann’,kennt,
                den,’Mann’,der,ihn,grüßt],[]).

⟨Parse-Aufruf ⟩≡

C (7) parse(s, [der,’Mann’,...], [])
C (8) headcorner(v, s,
         [der,’Mann’,...], [kennt,..],[..], [])
E (9) s--->nom, v, akk, rs

⟨Linkskontext nom von v aus (9) ⟩≡

C (11) parse(nom, [der,’Mann’,kennt,...],
                             [kennt,...])
C (12) headcorner(n(nom),nom,
          [der,’Mann’,kennt,...],
          [’Mann’,..],[..], [kennt,...])
E (13) ... E (13) nom--->det(nom), n(nom)
C ... E (15) parse(det(nom), [der,’Mann’,...],
                                 [’Mann’,...])
C E (13) headcorner(nom, nom,
            [der,’Mann’,kennt,...],
            [der,’Mann’,..],[..], [kennt, ...])
E (12) headcorner(n(nom),nom,
          [der,’Mann’,kennt,...],
          [’Mann’,..],[..], [kennt,...])
E (11) parse(nom, [der,’Mann’,kennt,...],
                             [kennt,...])

Beispiel zur Head-Corner-Erkennung (Forts.)

⟨Teil akk des Rechtskontexts akk,rs von v aus (9) ⟩≡

C (10) parse(akk, [den,’Mann’,der,ihn,grüßt],_G933)
... (analog)
E (10) parse(akk, [den,’Mann’,der,ihn,grüßt],
                             [der,ihn,grüßt])

⟨Teil rs des Rechtskontexts akk,rs von v aus (9) ⟩≡

C (11) parse(rs, [der, ihn, grüßt], _G972)
C (12) headcorner(v, rs,
          [der,ihn,grüßt], [grüßt],[], [])
E (13) ...
E (13) rs--->rnom, akk, v

⟨Teil akk des Linkskontexts rnom,akk von v aus (13) ⟩≡

C (15) parse(akk, [der,ihn,grüßt], [grüßt])
F ...
C (15) parse(akk, [ihn,grüßt], [grüßt])
E (15)

⟨Teil rnom des Linkskontexts rnom,akk von v aus (13) ⟩≡

C (16) parse(rnom, [der,ihn,grüßt], [ihn,grüßt])
C (17) headcorner(rnom, rnom,
         [der,ihn,grüßt],[der,..],[..],[ihn,grüßt])
E (17)
E (16) parse(rnom, [der,ihn,grüßt], [ihn,grüßt])

⟨Teil rs des Rechtskontexts akk,rs von v aus (9) ⟩+≡

C (13) headcorner(rs, rs,
       [der,ihn,grüßt], [der,ihn,grüßt],[], [])
E (13)

Beispiel zur Head-Corner-Erkennung (Ende)

⟨Rückgabe der Ergebnisse⟩≡

 E (12) headcorner(v, rs,
                   [der,ihn,grüßt], [grüßt],[], [])
 E (11) parse(rs, [der,ihn,grüßt], [])

 C (9) E (9) headcorner(s, s,
        [der,’Mann’,...], [der,’Mann’,...],[], [])
 E (8) headcorner(v, s,
        [der,’Mann’,kennt,...], [kennt,...],[...], [])
 E (7) parse(s,
       [der,’Mann’,kennt,den,’Mann’,der,ihn,grüßt],
       [])
Yes b) Head-Corner-Parser in Prolog

(für DCGs, mit Rücksetzen)

In den Erkenner bauen wir eine Ausgabe der Syntaxbäume ein.

leaf/5 gibt im ersten Argument einen Baum aus:

⟨headcorner.pl⟩≡

  parse(Ws,Tree) :-
      start_symbol(S), parse(S,Ws,[],Tree).
  parse(A,I,J,Tree) :-
      (head(B,A) ; B=A),
      leaf([B|Trees],I,I2,J2,J),
      headcorner([B|Trees],A,I,I2,J2,J,Tree).
  leaf([A|Trees],I,I2,J2,J) :-
      ( (A ---> []),
          I=I2, I2=J2, J2=J, Trees=[]
      ; append(_,I2,I), I2=[W|J2], append(_,J,J2),
          (word(W,A), Trees=[W] ; A=[W], Trees=[])
      ).

headcorner/7 setzt eine Kopfkonstituente nach oben fort:

⟨headcorner.pl⟩+≡

headcorner([B|BTs],A,I,I2,J2,J,Tree) :-
    ( I=I2, J2=J, B=A, Tree=[B|BTs] % A
    ; (C ---> Gamma), % / \
      head((C ---> Gamma),Alpha,B,Beta), % / ? \
      reverse(Alpha,AlphaRev), % i-?- C -?-j
      parse_list_rev(AlphaRev,I,I3,I2,AlphaRs),%/ | \
      reverse(AlphaRs,AlphaTs), % Alpha B Beta
      parse_list(Beta,J2,J3,BetaTs), % / | / \ | \
      append(AlphaTs,[[B|BTs]|BetaTs],Ts),%i3-i2 BTs j2-j3
      headcorner([C|Ts],A,I,I3,J3,J,Tree)
    ).

Die Grammatik bestimme die Hauptkonstituente(n) einer Regel wieder grob durch head/2 allein aus der Phrasenkategorie:

⟨headcorner.pl⟩+≡

head((C ---> Gamma),Alpha,B,Beta) :-
    roots(Gamma,GammaList),
    head(B,C),
    append(Alpha,[B|Beta],GammaList).

Bei parse list und parse list rev gibt man jetzt eine Liste von Bäumen aus:

⟨headcorner.pl⟩+≡

parse_list_rev([],_,J,J,[]).
parse_list_rev([A|As],I,K,J,[Tree|Trees]) :-
    parse_rev(A,I,L,J,Tree),
    parse_list_rev(As,I,K,L,Trees).

parse_rev(A,I,K,J,Tree) :-
    append(_,K,I), append(_,J,K),
    parse(A,K,J,Tree).

parse_list([B|Bs],I,J,[Tree|Trees]) :-
    parse(B,I,K,Tree), parse_list(Bs,K,J,Trees).
parse_list([],I,J,[]) :- I = J.

⟨roots.pl⟩

Man könnte mit einem besseren roots/2 noch die Grammatikregeln nach gemeinsamen Kopfkonstituenten faktorisieren.

Beispiel zur Head-Corner-Analyse

Die Reihenfolge der Verwendung lexikalischer Elemente ist:

⟨Beispiel⟩≡

?- [’cfg.hc.grm’, headcorner], trace(leaf,[exit]).
?- parse(s,[der,’Mann’,kennt,den,’Mann’,der,ihn,grüßt],
         [],Tree), printTree(Tree,2).
E (9) leaf([v,kennt],[der,..],[kennt,..],[den,..],[])
E (13) leaf([n(nom),’Mann’],[der,..],[’Mann’..],
                            [kennt,..],[kennt,..])
E (17) leaf([det(nom),der],[der,..],[der,..],
                          [’Mann’,..],[’Mann’,kennt,..])
E (12) leaf([n(akk),’Mann’],[den,..],[’Mann’,der,..],
                            [der,ihn,..],[der,..])
E (16) leaf([det(akk),den],[den,..],[den,..],
                            [’Mann’,der,..],[’Mann’,..])
E (13) leaf([v,grüßt],[der,ihn,grüßt],[grüßt],[],[])
E (17) leaf([akk,ihn],[der,ihn,..],[ihn,..],[g..],[g..])
E (17) leaf([akk,ihn],[ihn,g..],[ihn,g..],[g..],[g..])
E (18) leaf([rnom,der],[der,ihn,..],[der,..],
                            [ihn,..],[ihn,..])
     s
       nom
         det(nom) der 3.
         n(nom) ’Mann’ 2.
       v kennt 1.
       akk
         det(akk) den 5.
         n(akk) ’Mann’ 4.
       rs
         rnom der 8.
         akk ihn 7.
         v grüßt 6.

TABELLENVERFAHREN (CHART PARSER)

Bei den Analyseverfahren mit Rücksetzungen wird ein Teilausdruck bei einer Rücksetzung i.a. nochmals analysiert. Das vermeidet man bei den Tabellenverfahren, indem man die Analysen von Teilausdrücken in einer Tabelle speichert und bei Bedarf dort nachschlägt.

Braucht man nur ein Erkennungsverfahren, so speichert man in der Tabelle T zu jedem Teilausdruck v der Eingabe w alle Kategorien A, die v hat, d.h. T (v) = { A | A ⇒^∗ v }.

1. Der Erkenner von Cocke, Younger, Kasami (CYK)

Vor.: Kontextfreie Grammatik G = (N, Σ, S, P ) in Chomsky- Normalform: alle Regeln haben die Form (A → a) oder (A → BC) mit A, B, C ∈ N, a ∈ Σ.

Eingabe: w = a_1 · · · a_n ∈ Σ^+.

Methode: Konstruiere für die Positionen 0 ≤ i < j ≤ n alle Kanten i^A−→ j nach den Regeln:

a = a_{i+1}, (A → a) ∈ P i^A−→ i + 1

(Scan)

i^B−→ k, k^C−→ j, (A → BC) ∈ P i^A−→ j

(Complete)

Ausgabe: 0^S−→ n wird konstruiert ⇐⇒ w ∈ L(G).

Das folgt aus: i^A−→ j wird konstruiert ⇐⇒ A ⇒^∗ a_{i+1} · · · a_j .

Da es nur endlich viele Kanten i^A−→ j zu w gibt, terminiert das Verfahren (bei geeigneter Verwaltung).

Das CYK-Verfahren ist ein „bottom-up“-Verfahren: die Konstruktion der Kanten beginnt bei den i^{ai+1}−→ i + 1 und nutzt S nicht aus.

Programmschleife zur Konstruktion der Kanten

Die Kanten stellt man als Eintrag einer n × n-Tabelle T dar:

T (i, j) = { A | i^A−→ j } = { A | A ⇒^∗ a_{i+1} · · · a_j }.

Man kann die Kanten in verschiedener Reihenfolge bilden, z.B. zuerst alle Kanten i^A−→ k mit i < k < j, dann alle j − 1^A−→ j, dann alle j − 2^A−→ j, bis man alle mit i < k ≤ j hat, usw.:

for j = 1 to n do

T (j − 1, j) := { A ∈ N | (A → a_j ) ∈ P }; if j > 1 then

for i = j − 2 to 0 do

T (i, j) := ⋃ i<k<j { A ∈ N | B ∈ T (i, k), C ∈ T (k, j),

(A → BC) ∈ P } od od Aufgabe: Konstruiere die Kanten in der Reihenfolge: zuerst alle Kanten der Länge 1, die i^A−→ i + 1, dann die der Länge 2, die i^A−→ i + 2, usw.

Komplexität der Erkennung mit CYK

• Für i = 0, . . . , n − 1, initialisiere Feld T (i, i + 1). Dazu muß für jede Regel geprüft werden, ob sie von der Form A → a_i ist. Das erfordert jeweils O(|P |) Schritte. Insgesamt also O(|w| ∗ |P |) ≤ O(|w| ∗ |G|) Schritte.

• Angenommen, T (i, j) ist für alle Teilwörter a_{i+1} . . . a_j der Länge j − i < k (endgültig) bestimmt.

Zum Auffüllen der T (i^′, j^′) mit |j^′ − i^′| = k ist für alle Regeln A → BC und alle k^′ mit i^′ < k^′ < j^′ zu prüfen, ob B ∈ T (i^′, k^′) und C ∈ T (k^′, j^′). Um T (i^′, j^′) zu füllen, braucht man

O(|P | × (j^′ − i^′) × |N|)

viele Schritte (auf einer RAM), wobei höchstens |N| Schritte für die Suche durch das Feld T (i, j) nötig sind.

• Um alle Felder von T zu füllen, braucht man also insgesamt höchstens

O(|w| ∗ |G|) + ∑ i<j≤|w|

O(|P | × (j − i) × |N|)

= O(|w| ∗ |G|) + O(|P | × |N| × ∑ j≤|w|

j(j + 1) 2 )

≤ O(|P | × |N| ×

|w|^3 3 ) = O(|w|^3 × |G|^2)

Für Regeln der Form (A → B_1 · · · B_k) ist der Aufwand größer!

CYK-Erkenner in Prolog

Eine Kante i^A−→ j stellen wir als Prolog-Fakt kante(I,A,J) dar. Wir füllen die Tabelle unter der Annahme, daß alle Kanten mit Indizes < j eingetragen sind, indem der Reihe nach die Felder T (j − 1, j), . . . , T (0, j) gefüllt werden.

⟨cyk-rec.cfg.pl⟩≡

% Cocke/Younger/Kasami-Erkenner

parse([W|Ws]) :-
    retractall(kante(_,_,_)), % alte Tabelle loeschen
    parse(0,[W|Ws]).

parse(I,[W|Ws]) :- % on-line Vorgehen
    scan(I,W),
    J is I+1,
    complete(I,J), %
    parse(J,Ws).
parse(J,[]) :-
    kante(0,A,J),
    start_symbol(A).

scan(I,W) :-
    ((A ---> [W]) ; word(W,A)), % Lexikoneintrag
    J is I+1,
    add(kante(I,A,J)),
    fail. % weitere Lexikoneintraege zu W ?
scan(_,_).

Eine Kante wird nur ergänzt, falls sie keine Instanz einer vorhandenen Kante ist. Dazu ersetzt man in der Kante die Variablen durch c(0),c(1),... und versucht, diese variablenfreie

Variante zu beweisen; geht das, ist die Kante die Instanz einer vorhandenen. Dann macht man die Belegung von Variablen rückgängig; sonst speichert man die Kante:

⟨cyk-rec.cfg.pl⟩+≡

add(Fakt) :-
    (\+ \+ (numbervars(Fakt,c,0,_), Fakt)
    -> true ; % nichts tun, falls Fakt[c/X] beweisbar
        assert(Fakt)
    ).

Nachdem Feld (J-1,J) gefüllt ist, werden die Felder (K,J) mit K < J-1 gefüllt:

⟨cyk-rec.cfg.pl⟩+≡

complete(K,J) :-
    (K = 0 -> true
    ; I is K-1, complete(I,K,J), complete(I,J)
    ).

complete(I,K,J) :- % I < K < J
    kante(I,B,K),
    kante(K,C,J),
    (A ---> B,C),
    add(kante(I,A,J)),
    fail.
complete(I,K,J) :-
    (I = 0 -> true
    ; L is I-1, complete(L,K,J)
    ).

Beispiel zur Erkennung mit CYK

⟨Beispiel mit cfg.grm⟩≡

?- [’cfg.grm’, ’cyk-rec.cfg.pl’].
?- trace([complete/2,complete/3],[call,exit]).
?- parse([a,b,a]), listing(kante).
 T Call: (9) complete(0, 1)
 T Exit: (9) complete(0, 1)
 T Call: (10) complete(1, 2)
 T Call: (11) complete(0, 1, 2)
 T Exit: (11) complete(0, 1, 2)
 T Call: (11) complete(0, 2)
 T Exit: (11) complete(0, 2)
 T Exit: (10) complete(1, 2)
 T Call: (11) complete(2, 3)
 T Call: (12) complete(1, 2, 3)
 T Call: (13) complete(0, 2, 3)
 T Exit: (13) complete(0, 2, 3)
 T Exit: (12) complete(1, 2, 3)
 T Call: (12) complete(1, 3)
 T Call: (13) complete(0, 1, 3)
 T Exit: (13) complete(0, 1, 3)
 T Call: (13) complete(0, 3)
 T Exit: (13) complete(0, 3)
 T Exit: (12) complete(1, 3)
 T Exit: (11) complete(2, 3)
:- dynamic kante/3.
kante(0, a(sg), 1).
kante(1, b(sg), 2).
kante(0, s(sg), 2).
kante(2, a(sg), 3).
kante(1, b(pl), 3).
kante(0, s(pl), 3).

2. Der Erkenner von Earley

Das Verfahren von Earley kann man als eine Verbesserung des CYK-Verfahrens ansehen: die Grammatik braucht nicht in Chomsky-Normalform zu sein, und die Analyse benutzt die Kategorie S als „Top-Down-Filter“, d.h. für die Eingabe werden nur ihre Analysen als S-Ausdruck konstruiert.

Vor.: Beliebige kontextfreie Grammatik G = (N, Σ, S, P ) Eingabe: w = a_1 · · · a_n ∈ Σ^∗ Methode: Konstruiere für die Positionen 0 ≤ i ≤ j ≤ n alle Kanten i^{A→α·β}−→ j, wo (A → αβ) ∈ P , nach den Regeln: (S → α) ∈ P

0^{S→·α}−→ 0

(Predict)

i^{A→α·Cβ}

−→ j, (C → γ) ∈ P j^{C→·γ}−→ j

(Predict)

i^{A→α·aβ}

−→ j, a = a_{j+1} i^{A→αa·β}

−→ j + 1

(Scan)

i^{A→α·Cβ}

−→ k, k^{C→γ·}−→ j i^{A→αC·β}

−→ j

(Complete)

Ausgabe: Eine Kante der Form 0^{S→α·}−→ n wird erzeugt ⇐⇒ w ∈ L(G).

Zum Beweis zeigt man:

Beh.: Die Kante i^{A→α·β}−→ j wird genau dann konstruiert, wenn (i) S ⇒^∗ a_1 · · · a_iAγ für ein γ ∈ (N ∪ Σ)^∗, und (ii) α ⇒^∗ a_{i+1} · · · a_j .

Bew. =⇒ (Korrektheit der Regeln): Zeige (i), (ii) durch Induktion über die Anzahl der Schritte zur Erzeugung der Kante. ⇐= (Vollständigkeit der Regeln): Für (A → αβ) ∈ P gelte (i) und (ii). Dann gibt es Linksableitungen

S ⇒^{m1} a_1 · · · a_iAγ und α ⇒^{m2} a_{i+1} · · · a_j bestimmter Längen m_1, m_2. Durch Induktion über m_1 + m_2 zeigt man, daß i^{A→α·β}−→ j konstruiert wird.^1 Die Analyse von Earley ist eine Mischung aus Top-Down- und Bottom-Up-Vorgehen:

• (Predict) beginnt zu Erwartungskategorien C „top-down“ neue partielle Erkennungsversuche.

• (Complete) erweitert „bottom-up“ eine partielle erkannte A-Phrase durch eine vollständig erkannte C-Phrase.

Der Earley-Erkenner terminiert, da es zu jedem w = a_1 · · · a_n nur endlich viele Kanten i^{A→α·β}−→ j mit 0 ≤ i ≤ j ≤ n gibt.

1Man braucht die Hilfsbehauptung, daß für m_1 > 0 ein S ⇒^{m1−1} a_1 · · · a_kA^′γ^′ entsteht und alle k^{A′→·α′}−→ k erzeugt werden.

Earley-Erkenner in Prolog (für CFGs, mit Speichern einer Tabelle kante/5)

Das Lexikon kann mit word(W,Kat) abgetrennt sein.

⟨earley-rec.pl⟩≡

:- dynamic kante/5. % kante(I,J,A,Alpha,Beta)

erkenne(Wortfolge) :-
    retractall(kante(_I,_J,_A,_Alpha,_Beta)),
    predict(0),
    erkenne(0,Wortfolge).

erkenne(I,[Wort|Rest]) :-
    J is I + 1,
    scan(I,J,Wort),
    erkenne(J,Rest).
erkenne(I,[]) :- % Eingabe erkannt?
    start_symbol(S),
    kante(0,I,S,_Alpha,[]).

scan(I,J,Wort) :- % i -Wort-> j=i+1
    ( word(Wort,A),
        closure(I,J,A,[Wort],[])
    ; ( kante(K,I,A,Alpha,([Wort],Beta))
      ; kante(K,I,A,Alpha,[Wort]),Beta = []
      ), append(Alpha,[[Wort]],AlphaW),
        closure(K,J,A,AlphaW,Beta)
    ), fail.
scan(_,_,_).

Durch closure(I,J,A,Alpha,Beta) werden auf i^{A→α·β}−→ j rekursiv die Regeln (Complete) und (Predict) anwendet, bis keine weiteren Kanten hinzukommen:

Earley-Erkenner in Prolog (Forts.)

Entsteht eine neue Kante i^{B→β·}−→ j, so werden „rückwärts“ alle „wartenden“ Kanten k^{A→α·Bβ}−→ i damit verbunden (Complete):

⟨earley-rec.pl⟩+≡

closure(I,J,B,Beta,[]) :-
    (add(kante(I,J,B,Beta,[]))
    -> complete(I,J,B) ; true).

Entsteht eine neue Kante i^{A→α·Bβ}−→ j, so werden „vorwärts“ Kanten j^{B→·γ}−→ j in die Tabelle eingetragen und bei γ = ϵ mit vorhandenen verkettet (Predict):

⟨earley-rec.pl⟩+≡

closure(I,J,A,Alpha,(B,Bs)) :-
    !, add(kante(I,J,A,Alpha,(B,Bs))),
    predict(J,B).
closure(I,J,A,Alpha,B) :-
    add(kante(I,J,A,Alpha,B)),
    predict(J,B).

Nach der Verkettung zweier Kanten muß die Tabelle mit der neuen Kante weiter unter Verkettung abgeschlossen werden:

⟨earley-rec.pl⟩+≡

complete(I,J,B) :-
    ( kante(K,I,A,Alpha,(B,Beta))
    ; kante(K,I,A,Alpha,B), Beta = []
    ),
    append(Alpha,[B],AlphaB),
    closure(K,J,A,AlphaB,Beta),
    fail.
complete(_,_,_).

Earley-Erkenner in Prolog (Forts.)

Die Voraussage (Predict) sucht in der Grammatik nach den zur Erwartung S bzw. A passenden Regeln, trägt sie in die Tabelle ein und schließt diese unter Verkettung ab:

⟨earley-rec.pl⟩+≡

predict(0) :-
    start_symbol(S), predict(0,S), fail.
predict(0).

predict(J,A) :-
    (A ---> Alpha), closure(J,J,A,[],Alpha),
    fail.
predict(J,A) :-
    (A ---> []), % mit Loeschkanten verketten,
    complete(J,J,A), % auch wenn diese schon
    fail. % in der Tabelle sind!
predict(_,_).

Nur neue Kanten, die keine Instanz einer vorhandenen Kante sind, sollen in die Tabelle eingetragen werden:

⟨earley-rec.pl⟩+≡

add(Fact) :-
    ((var(Fact) ; known(Fact))
    -> fail ; assertz(Fact)).
known(Term) :-
    \+ \+ (numbervars(Term,c,0,_), Term).

Zu numbervars(Term,c,0,N) siehe den CYK-Erkenner.

Earley-Parser in Prolog (für CFGs, mit Speichern einer Tabelle kante/5)

Im erkannten Teil Alpha speichern wir jetzt eine Baumliste.

⟨earley.pl⟩≡

:- dynamic kante/5. % kante(I,J,A,Alpha,Beta)

parse(Wortfolge,Baum) :-
    retractall(kante(_I,_J,_A,_Alpha,_Beta)),
    predict(0),
    parse(0,Wortfolge,Baum).

parse(I,[Wort|Rest],Baum) :-
    J is I + 1,
    scan(I,J,Wort),
    parse(J,Rest,Baum).
parse(I,[],[S|Alpha]) :- % Eingabe erkannt?
    start_symbol(S),
    kante(0,I,S,Alpha,[]).

scan(I,J,Wort) :- % i -Wort-> j=i+1
    ( word(Wort,A),
        closure(I,J,A,[Wort],[])
    ; ( kante(K,I,A,Alpha,([Wort],Beta))
      ; kante(K,I,A,Alpha,[Wort]),Beta = []
      ), append(Alpha,[[Wort]],AlphaW),
        closure(K,J,A,AlphaW,Beta)
    ), fail.
scan(_,_,_).

Durch closure(I,J,A,Alpha,Beta) werden auf i^{A→α·β}−→ j rekursiv die Regeln (Complete) und (Predict) anwendet, bis keine weiteren Kanten hinzukommen:

Earley-Parser in Prolog (Forts.)

Entsteht eine neue Kante i^{B→β·}−→ j, so werden alle „wartenden“ Kanten k^{A→α·Bβ}−→ i damit verbunden (Complete):

⟨earley.pl⟩+≡

closure(I,J,B,Beta,[]) :-
    (add(kante(I,J,B,Beta,[]))
    -> complete(I,J,B) ; true).

Entsteht eine neue Kante i^{A→α·Bβ}−→ j, so muß mit (Predict) bei j jede B-Regel vorgeschlagen und ggf. B gelöscht werden:

⟨earley.pl⟩+≡

closure(I,J,A,Alpha,(B,Bs)) :-
    !, add(kante(I,J,A,Alpha,(B,Bs))),
    predict(J,B).
closure(I,J,A,Alpha,B) :-
    add(kante(I,J,A,Alpha,B)),
    predict(J,B).

Werden Kanten verkettet, muß weiter abgeschlossen werden:

⟨earley.pl⟩+≡

complete(I,J,B) :-
    ( kante(K,I,A,Alpha,(B,Beta))
    ; kante(K,I,A,Alpha,B), Beta = []
    ),
    kante(I,J,B,Trees,[]),
    append(Alpha,[[B|Trees]],AlphaB),
    closure(K,J,A,AlphaB,Beta),
    fail.
complete(_,_,_).

Earley-Parser in Prolog (Forts.)

Mit predict(j,A) werden bei j alle A-Regeln eingetragen und –für löschbare A– mit vorhandenen Kanten verkettet:

⟨earley.pl⟩+≡

predict(0) :-
    start_symbol(S),predict(0,S),fail.
predict(0).

predict(J,A) :-
    (A ---> Alpha),
    closure(J,J,A,[],Alpha),
    fail.
predict(J,A) :-
    (A ---> []), % mit Loeschkanten verketten,
    complete(J,J,A), % auch wenn diese schon
    fail. % in der Tabelle sind!
predict(_,_).

Nur neue Kanten, die keine Instanz einer vorhandenen Kante sind, sollen in die Tabelle eingetragen werden:

⟨earley.pl⟩+≡

add(Fact) :-
    ((var(Fact);known(Fact)) -> fail ; assertz(Fact)).
known(Term) :-
    \+ \+ (numbervars(Term,c,0,_), Term).

Der Earley-Parser kann divergieren, weil (Complete) evtl. (für i = k und α ⇒^∗ ϵ, β ⇒^∗ ϵ) einen Baum nach oben erweitert, ohne mehr von der Eingabe abzudecken.

Beispiel

⟨cfg2.grm⟩≡

start_symbol(s).
:- op(1100,xfx,’--->’).
s ---> ap, bp.
ap ---> [].
bp ---> s.
word(a,ap).
word(b,bp).

⟨Der Earley-Erkenner terminiert:⟩≡

?- [’cfg2.grm’,earley-rec],
   erkenne([b]).
Yes

⟨Der Earley-Parser divergiert:⟩≡

?- [’cfg2.grm’,earley], trace(add,[exit]),
   parse([b],S).

 E (12) add(kante(0, 0, s, [], (ap, bp)))
 E (14) add(kante(0, 0, ap, [], []))
 E (16) add(kante(0, 0, s, [[ap]], bp))
 E (18) add(kante(0, 0, bp, [], s))
 E (12) add(kante(0, 1, bp, [b], []))
 E (14) add(kante(0, 1, s, [[ap],
                            [bp,b]], []))
 E (16) add(kante(0, 1, bp, [[s,[ap],[bp,b]]], []))
 E (18) add(kante(0, 1, s, [[ap],
                            [bp,[s,[ap],[bp,b]]]], []))
...

(Bei einer solchen Eingabe divergiert natürlich jeder Parser.)

GRAMMATIKÜBERSETZUNG Um die Analyse zu beschleunigen, soll das Suchen nach einer Regel (A → α) ∈ P so geändert werden, daß man nicht immer alle Grammatikregeln durchläuft, sondern

(i) die Kategorien des nächsten Eingabesymbols und (ii) die für die Resteingabe „erwarteten“ Kategorien benutzt, um Teilmengen P^′ ⊆ P zu bestimmen, die alle in der jeweiligen Analysesituation noch nötigen Regeln umfassen.

Hilfsrelationen zwischen Kategorien Aus den DCG-Regeln der Grammatik berechnen wir (i) die löschbaren Kategorien {A | A ⇒^+ ϵ}, (ii) die Kettenrelation { (A, B) | A ⇒^+ B }, (iii) die Linke-Ecken-Relation

<_{lc} := {(B, A) | (A → αBγ) ∈ P, α ⇒^∗ ϵ} und deren reflexive, transitive Hülle <^∗_{lc} := {(B, A) | ∃γ ∈ Cat^∗ A ⇒^∗ Bγ} (iv) ggf. die Hauptkategorie-Relation <_{hc} := {(B, A) | (A → αBγ) ∈ P } und ihre reflexive, transitive Hülle <^∗_{hc}.

Wie man diese Regeln benutzt, zeigen wir an der Left-Corner- Analyse mit Rücksetzungen und an den Tabellenverfahren von Earley und dem für die Left-Corner-Analyse.

Nichtdeterministische Left-Corner-Analyse für übersetzte Grammatiken

Ist E die für ein Präfix der Resteingabe „erwartete“ Kategorie, D die Kategorie eines nicht-leeren, schon erkannten Anfangsstücks der Resteingabe und b das auf dieses Anfangsstück folgende Eingabesymbol (bzw. seine Kategorie), so sei predict_{lc}(E, D, c) :=

{A → α · Bβ_1 · Cβ_2) ∈ P̈ | A <^∗_{lc} E, αβ_1 ⇒^∗ ϵ, B ⇒^∗ D, c <^∗_{lc} C} Diese Teilmenge von P̈ kann für jedes (E, D, b) unabhängig von der Eingabe vorberechnet werden.

Der frühere Left-Corner-Erkenner kann damit beschleunigt werden, indem man das Probieren einer passenden Regel auf die jeweils als brauchbar vorausgesagten beschränkt: Konfigurationen: (η, γ, w) ∈ Cat^∗ × (Cat ∪ P̈ )^∗ × Σ^∗ Anfang bzw. Ende: (S, ϵ, a_1 · · · a_n) bzw. (ϵ, S, ϵ) Eingabe lesen: (ηE, γ, aw) ⊢_a (ηE, γa, w), falls a <^∗_{lc} E. Partiell reduzieren: (ηE, γD, cw) ⊢ (ηEC, γ(A → αBβ_1 · Cβ_2), cw), falls (A → α · Bβ_1 · Cβ_2) ∈ predict_{lc}(E, D, c) Vervollständigen 1: (ηB, γ(A → α · Bβ_1Cβ_2)D, cw) ⊢ (ηC, γ(A → αBβ_1 · Cβ_2), cw) falls B ⇒^∗ D, β_1 ⇒^∗ ϵ, und c <^∗_{lc} C, Vervollständigen 2: (ηB, γ(A → α · Bβ)D, w) ⊢ (η, γA, w), falls B ⇒^∗ D und β ⇒^∗ ϵ

Tabellenparser für übersetzte Grammatiken Aus der Grammatik seien die Hilfsrelationen

(i) die löschbaren Kategorien {A | A ⇒^+ ϵ} ⊆ Cat, (ii) die Kettenrelation { (A, B) | A ⇒^+ B } ⊆ Cat^2, (iii) die Linke-Ecken-Relation

<_{lc} := {(B, A) | (A → αBγ) ∈ P, α ⇒^∗ ϵ} und deren reflexive, transitive Hülle

<^∗_{lc} := {(B, A) | ∃γ ∈ Cat^∗ A ⇒^∗ Bγ} (iv) und folgende Teilmengen von Ṗ , für E, D ∈ Cat, c ∈ Σ:

predict_{ep}(E, c) := {(A → α · Cβ) ∈ Ṗ | A <^∗_{lc} E, α ⇒^∗ ϵ, c <^∗_{lc} C} predict_{lc}(E, D, c) := {(A → αBβ · Cγ) ∈ Ṗ |

A <^∗_{lc} E, αβ ⇒^∗ ϵ, B ⇒^∗ D, c <^∗_{lc} C} vorberechnet. Wir benutzen sie nun, um

(i) die Suche nach passenden Grammatikregeln während der der Leftcorner- bzw. Earley-Erkennung einzuschränken, (ii) dafür zu sorgen, daß nur kanonische Lösch- und Kettenableitungen gesucht werden.

Aus (ii) erhalten wir ein für jede Eingabe terminierendes Analyseverfahren, wobei in der Tabelle die Kanten i^{A→α·β}−→ j durch α-Bäume zu partiellen Analysen erweitert werden.

Verallgemeinerter Earley-Erkenner Speichern von Erwartungen j · B, „bei Position j beginne eine B-Phrase“, ersparen das Suchen nach Kanten i^{A→α·Bβ}−→ j. Vor.: Kontextfreie Grammatik G = (N, Σ, S, P ),

Hilfsrelationen Cat ⇒^∗ ϵ, Cat ⇒^∗ Cat, <^∗_{lc}, predict_{ep} Eingabe: w = a_1 · · · a_n ∈ Σ^∗, Methode: Konstruiere für die Positionen 0 ≤ i ≤ j ≤ n alle Kanten i^{A→α·β}−→ j, wo (A → αβ) ∈ P , nach den Regeln:

start(S)

0 · S

(Start) i^{A→α·Bβ}

−→ j j · B

(Expect)

j · C, (A → α · Bβ) ∈ predict_{ep}(C, a_{j+1}) j^{A→α·Bβ}

−→ j

(Predict)

j = i + 1 i^{aj →aj ·}

−→ j

(Scan)

i^{A→α·Bβ}

−→ k, B ⇒^∗ D, k^{D→δ·}−→ j, β ⇒^∗ ϵ i^{A→αBβ·}

−→ j

(Complete_1)

i^{A→α·BβCγ}

−→ k, B ⇒^∗ D, k^{D→δ·}−→ j, β ⇒^∗ ϵ, j < n, a_{j+1} <^∗_{lc} C i^{A→αBβ·Cγ}

−→ j

(Complete_2)

Ausgabe: Eine Kante 0^{S→α·}−→ n wird erzeugt ⇐⇒ w ∈ L(G).

Verallgemeinerter Leftcorner-Erkenner Der Aufbau der Kanten zu w = a_1 · · · a_n erfolgt durch (i) Festlegen von Erwartungskategorien für die Resteingabe:

start(S)

0 · S

(Start) i^{A→α·Bβ}

−→ j j · B

(Expect)

(ii) Lesen der Eingabewörter: j = i + 1 i^{aj →aj ·}

−→ j

(Scan)

(iii) Voraussagen brauchbarer Regeln: i · E, i^{D→δ·}−→ j, (A → αBβ · Cγ) ∈ predict_{lc}(E, D, a_{j+1}) i^{A→αBβ·Cγ}

−→ j

(Predict)

(iv) Abschluß unter Verkettung von Kanten:

i^{A→α·Bβ}

−→ k, B ⇒^∗ D, k^{D→δ·}−→ j, β ⇒^∗ ϵ i^{A→αBβ·}

−→ j

(Complete_1)

i^{A→α·BβCγ}

−→ k, B ⇒^∗ D, k^{D→δ·}−→ j, β ⇒^∗ ϵ, j < n, a_{j+1} <^∗_{lc} C i^{A→αBβ·Cγ}

−→ j

(Complete_2)

Neben den Erwartungskategorien j · B wird das nächste Eingabewort und c <^∗_{lc} C zum Filtern der Regeln benutzt wird.

Beobachtungen

(i) Die Menge P der Grammatikregeln wird zur Parsezeit nicht mehr durchsucht; stattdessen nur die Teilmengen predict_{ep} bzw. predict_{lc} und die Tabelle der Kanten.

(ii) Die verallgemeinerten Earley- und Leftcorner-Erkenner unterscheiden sich nur in (Predict).

Der Leftcorner-Erkenner erzeugt nur i^{A→α·β}−→ j mit i < j, d.h. wo α einen nicht-leeren Teil der Eingabe erkennt.

(iii) In (Scan) werden ad-hoc-Regeln (a → a) verwendet, die in (D → δ) in (Complete) eingesetzt werden.

(iv) Zum Erkennen kann man Kanten i^{A→α·β}−→ j zu i^{A→·β}−→ j vereinfachen, da der α-Teil nicht benutzt wird.

(v) Eine Kante k^{D→δ·}−→ j wird nur für k < j erzeugt.

Erweiterung der verallgemeinerten Erkenner zu Earley- und Leftcorner-Tabellenparsern

(i) Die Kanten i^{A→α·β}−→ j stehen für partielle Analysen von a_{i+1} · · · a_j , d.h. sind Adressen für Baumfolgen mit Wurzeln α und Blättern a_{i+1} · · · a_j .

(ii) In (Complete)-Schritten werden diese Baumfolgen um einen Baum zu B ⇒^∗ D → δ ⇒^∗ a_{k+1} · · · a_j mit k < j erweitert.

(iii) In (Complete_1) wird für i < k ein echt verzweigender Knoten A erzeugt, für i = k muß man eine kanonische Ableitung A ⇒^∗ D aufbauen, also B ⇒^∗ D ignorieren.

(iv) (Complete_2) erzeugt nur echt verzweigende Knoten A, mit k < j < l in k^B−→ j und j^C−→ l für ein l.

(v) Jede vollständige Analyse A ⇒^∗ a_i . . . a_j ≠ ϵ beginnt daher mit einer Kettenableitung A ⇒^∗ D zu einem Blatt D (und i = j) oder einem Verzweigungsknoten D. Daraus folgt, daß nur kanonische Bäume aufgebaut werden, und daß der Parser terminiert.

Bem. Kanten i^{D→δ·}−→ j kann man zu i^D−→ j vereinfachen, d.h. man braucht die vollständigen Analysen nur nach der Wurzel zu unterscheiden.

Implementierung

Auf die Implementierung der Erreichbarkeitsrelationen und die Parser für übersetzte Grammatiken gehen wir nicht ein. (Vielleicht kommt ein Nachtrag für den CFG-Fall; für den DCG-Fall siehe die Doku zum Seminar Grammatikimplementierung.)

DETERMINISTISCHE ANALYSE

Nicht-deterministische Parser „probieren aus“, ob die Erweiterung einer partiellen Analyse mit einer Grammatikregel oder einem Eingabewort zur Erkennung führt; wenn nicht, wird die Erweiterung rückgängig gemacht und eine Alternative gesucht. Deterministische Parser sind solche, bei denen in jeder Situation höchstens eine Operation (Regelanwendung oder Lesen) zur Erkennung führt. Sie existieren nur für Grammatiken (von Programmier- oder logischen Sprachen), für die jeder Satz nur eine Analyse hat.

Für jede erreichbare Konfiguration (γ, v) ist die nächste Operation eindeutig durch γ und die k-Vorausschau_k(v) bestimmt. Nondet. Transitionen (der Erkenner) Det.

Top-Down (S, w) ⊢^∗ (ϵ, ϵ) LL(k) (read) (aγ, av) ⊢_a (γ, v) (expand) (Bγ, v) ⊢_{(B→β)} (βγ, v) Leftcorner (ϵ, w) ⊢^∗ (S, ϵ) LC(k) (shift) (γ, av) ⊢_a (γa, v) (predict) (γB, v) ⊢_{(A→Bβ)} (γ(A → B · β), v) (compl-1) (γ(A → α · Bβ)B, v) ⊢_1 (γ(A → αB · β), v) (compl-2) (γ(A → α·), v) ⊢_2 (γA, v) Bottom-Up (ϵ, w) ⊢^∗ (S, ϵ) LR(k) (shift) (γ, av) ⊢_a (γa, v) (reduce) (αβ, v) ⊢_{(B→β)} (αB, v)

Vergleich der erkannten Sprachklassen Sei ein festes Alphabet gegeben und LL(k) die Menge der Sprachen über diesem Alphabet, die eine Grammtik hat, für die der Top-Down-Parser mit k-Vorausschau determinisitisch ist. Entsprechend für Leftcorner und LC(k) bzw. und Bottom-Up- Parser und LR(k). Man kann zeigen:

(i) LL(k) ⊂ LL(k + 1), (ii) LC(k) ⊂ LC(k + 1), (iii) LR(k) ⊂ LR(k + 1), (iv) LL(k) ⊆ LR(k), Für Programmiersprachen gibt es Parsergeneratoren, die Parser nach den LL- und LR-Verfahren erzeugen:

(i) YACC (oder Gnu/Bison) erzeugt LR(1)-Parser (ii) ANTLR erzeugt LL(1)-Parser Die praktisch wichtigste Klasse ist LR(1).

LR(k)-Analyse Für alle (ϵ, w) ⊢^∗ (γ, uv) mit u =_k(uv) soll es höchstens eine direkte Nachfolgekonfiguration (γ, u) ⊢ (γ̃, ũ) geben, die zur Erkennung irgendeiner Fortsetzung führt: (γ, uw̃) ⊢ (γ̃, ũw̃) ⊢^∗ (S, ϵ) Voraussetzung: G = (N, Σ, P, S) ist eine reduzierte kontextfreie Grammatik, also jedes Nonterminal A vom Startsymbol erreichbar und ∅ ≠ L(A).

LR(0)-Analyse Im Fall k = 0 ist die nächste Operation allein durch den Kellerinhalt bestimmt, also unabhängig von der Resteingabe. Für jede erreichbare Konfiguration (γ, v) muß es also mindestens ein w̃ ∈ Σ^∗ geben, das zur Erkennung führt, (γ, w̃) ⊢_? (γ^′, w̃^′) ⊢^∗ (S, ϵ) und die Operation ? bei ⊢_? hängt nur von γ ab. Wieviel müssen wir vom Keller γ = αβ wissen, um sicher zu sein, daß höchstens eine Reduktion mit der Regel (B → β) ∈ P in Frage kommt? Es kann dann nur eine Rechtsableitung S ⇒^∗_{rm} αBw̃ ⇒_{rm} αβw̃ = γw̃ geben, weshalb man die reduzierten Linkskontexte von B bzgl. S, rlc(B) := { α ∈ Cat^∗ | S ⇒^∗_{rm} αBw̃ für ein w̃ ∈ Σ^∗ }, kennen muß. Nun überlegt man sich leicht:

Prop. 12 Für jede reduzierte kontextfreie Grammatik gilt:

(i) ist (S → αBβ) ∈ P , so ist α ∈ rlc(B). (ii) ist (A → αBβ) ∈ P , so ist rlc(A) · α ⊆ rlc(B), (iii) die Mengen rlc(B) sind die kleinsten solchen Mengen. Folgerung: die rlc(B) sind reguläre Mengen über Cat, also kann ein endlicher Automaten überprüfen, ob α ∈ rlc(B).

Der LR(0)-Automat zu G Seien S^′, ▷, ◁ neue Symbole, P^′ := P ∪ {S^′ → ▷S◁} und Ṗ := { (A → α · β) | (A → αβ) ∈ P^′ }.

Zu q ⊆ Ṗ sei q das kleinste p ⊆ Ṗ mit

q ∪ { (B → ·β) | (A → α · Bγ) ∈ p, (B → β) ∈ P } ⊆ p. Definiere den det. endl. Automaten A_G = (Q, Σ^′, δ, I, F ) durch • I := {S^′ → ▷ · S◁}, • F := δ(I, S), (= das q ∈ Q mit (S^′ → ▷S · ◁) ∈ q) • δ(q, X) := { (A → αX · β) | (A → α · Xβ) ∈ q̄ }, • Q ⊆ 2^Ṗ ist die kleinste Teilmenge von 2^Ṗ , die I enthält und mit einem q auch alle nicht-leeren δ(q, X).

Übliche Namen: closure(q) für q, goto(q, X) für δ(q, X).

Hat man in A_G einen Reihe von Übergängen

q_0^{X1}−→ q_1^{X2}−→ q_2 . . . q_{m−1}^{Xm}−→ q_m d.h. q_i = δ(q_{i−1}, X_i) so gilt: (B → X_1 · · · X_m·) ∈ q_m =⇒ (B → ·X_1 · · · X_m) ∈ q_0.

Das ist klar für m = 0, und für m ≥ k > 0 ist (B → X_1 · · · X_k · X_{k+1} · · · X_m) genau dann in q_k, wenn es in q_k ist.

Ist γ = αβ = αX_1 · · · X_m der Keller einer Konfiguration (γ, v), so ergibt eine Reduktion mit (B → X_1 · · · X_m) (γ, w̃) = (αβ, w̃) ⊢_{(B→β)} (αB, w̃).

Die Kellerinhalte werden durch Lesen mit A_G in Zustände von Q überführt. Ist nun q_0 = δ(I, α), I^α−→ q_0^{X1}−→ q_1^{X2}−→ q_2 . . . q_{m−1}^{Xm}−→ q_m so kommt A_G nach einer Reduktion mit (B → β) durch Lesen des neuen Kellerinhalts in den Zustand I^α−→ q_0^B−→ δ(q_0, B).

Man definiert daher einen Kellerautomaten M_G, der Zustände von A_G auf seinem Keller stapelt, und folgende Übergänge hat:

(i) für p := δ(q, a), a ∈ Σ, q ∈ Q: q ⊢^M_a q p (ii) für (B → β) ∈ P , β = X_1 · · · X_m, q_0, . . . , q_m ∈ Q mit (B → ·β) ∈ q_0 und (B → X_1 · · · X_k · X_{k+1} · · · X_m) ∈ q_k für 0 < k ≤ m, und p = δ(q_0, B): q_0q_1 · · · q_m ⊢^M_{(B→β)} q_0 p.

Falls G eine LR(0)-Grammatik ist, ist M_G deterministisch: das oberste Kellersymbol q legt fest, was man tun muß:

(i) wenn es ein (B → X_1 · · · X_k · aX_{k+2} · X_m) ∈ q gibt und a das nächste Eingabesymbol ist, (i) anwenden, (ii) wenn es (B → X_1 · · · X_m·) ∈ q = q_m gibt, (ii) anwenden. Beh. Für α ∈ Cat^∗ gilt (mit (B → ·ϵ) = (B → ϵ·)) α ∈ rlc(B) ⇐⇒ es gibt (B → β) ∈ P mit (B → ·β) ∈ δ(I, α)

Parsing, H.Leiß, 3 Std., WS 2006/07 96

Beispiel (Appel, Modern Compiler Implementation, 1997, korrig.)

Grammatik G: 1 T → ( E ) 3 E → T 2 T → a 4 E → E + T

Automat A_G: Kellerautomat M_G:

action goto ( ) a + $ T E 0 s2 s1 4 1 r2 r2 r2 r2 r2 2 s2 s1 6 4 3 ok 4 s5 s7 5 r1 r1 r1 r1 r1 6 r3 r3 r3 r3 r3 7 s2 s1 8 8 r4 r4 r4 r4 r4

Eintrag bei Feld (q, X) je nach q^X−→ p:

sp shift p auf den Keller ri reduce mit Regel i = (B → β·) ∈ q ok accept Reduzieren mit (B → β·) :=

pop^{|β|} ; push(goto(top, B))

Erkennnungsbeispiel:

0, (a + a)$ ⊢ 02, a + a)$ ⊢ 021, +a$ ⊢ 026, +a$ ⊢ 024, +a$ ⊢ 0247, a$ ⊢ . . .

Implementierung des LR(0)-Automaten

Unterscheide Zustände I ⊆ Ṗ als (hoffentlich kleine) Teilmengen ihres Abschlusses I unter der Regel

(A → α · Bγ) ∈ I (B → β) ∈ P (B → ·β) ∈ I

(pr).

Die Zustände werden numeriert und durch Fakten state(N,R) dargestellt; die Abschlüsse durch Fakten closure(N,R).

⟨lr0-automat.pl⟩≡

:- dynamic changed/1, goto/3, done/1, action/3.

compute_lr0_automaton :-
    remove_old_automata,
    compute_initial_state, % endl.Automat A_G
    compute_successor_states, % endl.Automat A_G
    compute_actions. % Kellerautomat M_G

remove_old_automata :- % aufraeumen
    retractall(action(_,_,_)),
    retractall(maxstate(_)),
    retractall(state(_)),
    retractall(state(_,_)),
    retractall(closure(_,_)),
    retractall(goto(_,_,_)),
    retractall(changed(_)),
    retractall(done(_)).

Den Anfangszustand 0 legen wir mit neuen Symbolen ’start’ und ’end’ fest. (Da der Automat unabhängig von Eingaben berechnet wird, muß man mit ’end’ das Eingabeende simulieren.)

Anfangszustand und Abschluß

⟨lr0-automat.pl⟩+≡

compute_initial_state :-
    assert(state(0)),
    assert(maxstate(0)),
    ( start_symbol(S),
        add(state(0,(start --> [],[S,end]))),
        fail
    ; true ),
    close_state(0).

Abschluß I: ergänze zu den Regeln (A → α·Cβ) von I so lange die Regeln (C → ·γ), bis keine neuen mehr hinzukommen.

⟨lr0-automat.pl⟩+≡

close_state(P) :-
    ( state(P,Rule),
        add(closure(P,Rule)),
        add(changed(P)),
        fail
    ; (changed(P) -> close_state2(P) ; true)
    ).
close_state2(P) :-
    retract(changed(P)),
    ( closure(P,(A --> Alpha,[C|Beta])),
        (C --> GammaTuple),
        tupleToList(GammaTuple,Gamma),
        \+ known_instance(closure(P,(C --> [],Gamma))),
        add(closure(P,(C --> [],Gamma))),
        add(changed(P)),
        fail
    ; (changed(P) -> close_state2(P) ; true) ).

⟨lr0-automat.pl⟩+≡

% ---- Hilfsdefinitionen ----
add(A) :-
    known_instance(A), !.
add(A) :-
    assertz(A).
known_instance(A) :- % ist A[Xi/consti] beweisbar?
        \+ \+ (numbervars(A, const, 0, B), A).

tupleToList((A, B), [A|C]) :-
    !, tupleToList(B, C).
tupleToList(A, [A]).
% ----------------------------

Nachfolgerzustände

Nachdem ein Zustand Q berechnet ist, werden seine direkten Nachfolger bestimmt und Q wird mit done(Q) als erledigt markiert. Ist ein Nachfolger ein neuer Zustand, so werden auch dessen direkte Nachfolger berechnet, bis keine neuen Zustände mehr entstehen:

⟨lr0-automat.pl⟩+≡

compute_successor_states :-
    state(P), \+ done(P),
    !, compute_successors(P), add(done(P)),
    compute_successor_states.
compute_successor_states.

Um die Nachfolger p eines Zustands q zu bestimmen, brauchen wir alle B, für die es eine Regel (A → α·Bβ) ∈ q gibt. Für jedes solche B wird ein B-Nachfolger von p gebildet und ggf. mit einem schon vorhandenen Zustand identifiziert:

⟨lr0-automat.pl⟩+≡

compute_successors(P) :-
    ( setof(B,A^Alpha^Beta^closure(P,
             (A --> Alpha,[B|Beta])),Bs)
    -> compute_successors(P,Bs)
    ; true ).

compute_successors(P,[B|Bs]) :-
    mk_succ(P,B),
    compute_successors(P,Bs).
compute_successors(P,[]).

mk_succ(P,end) :- !.
mk_succ(P,B) :-
    ( closure(P,(A --> Alpha,[B|Beta])),
        append(Alpha,[B],AlphaB),
        add(state(newstate,(A --> AlphaB,Beta))),
        fail % omit: state(newstate)!
    ; close_state(newstate) ),
    number(newstate,Q),
    ( retract(state(newstate,Rule)),
        add(state(Q,Rule)),
        fail
    ; retract(closure(newstate,Rule)),
        add(closure(Q,Rule)),
        fail
    ; true ),
    add(goto(P,B,Q)).

Vergleich mit vorhandenen Zuständen

Ob der mögliche neue Zustand aus den geteilten Regeln R mit

closure(newstate,R) ein schon bekannter Zustand ist, wird

durch Vergleichen mit den bisher berechneten Zuständen ermittelt. Wenn ja, wird für die Übergänge die Nummer des bekannten Zustands benutzt, sonst die um 1 erhöhte bisherige maximale Zustandsnummer.

⟨lr0-automat.pl⟩+≡

number(newstate,N) :-
    setof(Q,state(Q),Qs),
    number(newstate,Qs,N).

number(newstate,[Q|Qs],N) :-
    ( same_states(newstate,Q)
    -> N=Q ; number(newstate,Qs,N) ).
number(newstate,[],N) :-
    maxstate(M),
    retract(maxstate(M)),
    N is M+1,
    assert(maxstate(N)),
    add(state(N)).

Da set of/3 die Ergebnisliste sortiert, lassen sich Zustände leicht vergleichen:

⟨lr0-automat.pl⟩+≡

same_states(Q,P) :-
    setof(Rule,closure(Q,Rule),ClosureQ),
    setof(Rule,closure(P,Rule),ClosureP),
    ClosureQ = ClosureP.

Berechnung der Aktionstabelle

Schließlich muß noch die Aktionstabelle des Kellerautomaten konstruiert werden.

⟨lr0-automat.pl⟩+≡

compute_actions :-
    state(Q),
    terminal(B),
    ( closure(Q,(A --> Alpha,[])),
        add(action(Q,B,reduce(A --> Alpha)))
    ; closure(Q,(_A --> _Alpha,[B|_])),
        goto(Q,B,P),
        add(action(Q,B,shift(P)))
    ; closure(Q,(A --> Alpha,[end])),
        add(action(Q,end,accept))
    ),
    fail.
compute_actions :-
    state(Q),
    terminal(B),
    (action(Q,B,_) -> true ; add(action(Q,B,error))),
    fail.
compute_actions.

terminal(B) :-
    (_ --> GammaTuple),
    tupleToList(GammaTuple,Gamma),
    member(B,Gamma),
    \+ (B --> _).

Der LR(0)-Parser (Kellerautomat)

Um eine Eingabe mit dem Kellerautomaten zu erkennen, brauchen wir noch die Stapelveränderungen:

⟨lr0-rec.pl⟩≡

parse(Ws) :- parse([0],Ws).

parse([Q|Qs],[W|Ws]) :-
    action(Q,W,Action),
    ( Action = reduce(A --> Alpha),
        reduce([Q|Qs],(A --> Alpha),RedStack),
        parse(RedStack,[W|Ws])
    ; Action = shift(P), parse([P,Q|Qs],Ws)
    ; Action = error, write(’\n reject’)
    ).
parse([Q|Qs],[]) :-
    action(Q,end,Action),
    ( Action = reduce(A --> Alpha),
        reduce([Q|Qs],(A --> Alpha),RedStack),
        parse(RedStack,[])
    ; Action = error, write(’\n reject’)
    ; Action = accept, write(’\n accept’)
    ).
reduce(Stack,(A --> Alpha),RedStack) :-
    length(Alpha,N),
    length(Remove,N),
    append(Remove,[R|Rest],Stack),
    goto(R,A,P),
    RedStack = [P,R|Rest].

Der Parser braucht nur action/3 und von goto/3 die Übergänge mit Nonterminalen — nicht die Interna der Zustände.

Beispiel 1

Die übliche DCG-Regelexpansion wird mit term expansion/2 ausgeschaltet.

⟨Grammatiken/lr0.grm⟩≡

term_expansion((A --> B), (A --> B)).

start_symbol(t).
e --> t.
e --> e, +, t.
t --> ’(’, e, ’)’.
t --> a.

Die Berechnung des LR(0)-Automaten ergibt:

⟨Berechnung des Automaten⟩≡

?- compute_lr0_automaton, listing(state/2),
   listing(closure), listing(goto), listing(action).

state(0, (start-->[], [t, end])).
state(1, (t-->[’(’], [e, ’)’])).
state(2, (t-->[a], [])).
state(3, (start-->[t], [end])).
state(4, (t-->[’(’, e], [’)’])).
state(4, (e-->[e], [+, t])).
state(5, (e-->[t], [])).
state(6, (t-->[’(’, e, ’)’], [])).
state(7, (e-->[e, +], [t])).
state(8, (e-->[e, +, t], [])).

closure(0, (start-->[], [t, end])).
closure(0, (t-->[], [’(’, e, ’)’])).
closure(0, (t-->[], [a])).

closure(1, (t-->[’(’], [e, ’)’])).
closure(1, (e-->[], [t])).
closure(1, (e-->[], [e, +, t])).
closure(1, (t-->[], [’(’, e, ’)’])).
closure(1, (t-->[], [a])).
closure(2, (t-->[a], [])).
closure(3, (start-->[t], [end])).
closure(4, (t-->[’(’, e], [’)’])).
closure(4, (e-->[e], [+, t])).
closure(5, (e-->[t], [])).
closure(6, (t-->[’(’, e, ’)’], [])).
closure(7, (e-->[e, +], [t])).
closure(7, (t-->[], [’(’, e, ’)’])).
closure(7, (t-->[], [a])).
closure(8, (e-->[e, +, t], [])).

goto(0, ’(’, 1).
goto(0, a, 2).
goto(0, t, 3).
goto(1, ’(’, 1).
goto(1, a, 2).
goto(1, e, 4).
goto(1, t, 5).
goto(4, ’)’, 6).
goto(4, +, 7).
goto(7, ’(’, 1).
goto(7, a, 2).
goto(7, t, 8).

action(0, ’(’, shift(1)).
action(0, a, shift(2)).
action(1, ’(’, shift(1)).
action(1, a, shift(2)).

action(2, +, reduce((t-->[a]))).
action(2, ’(’, reduce((t-->[a]))).
action(2, ’)’, reduce((t-->[a]))).
action(2, a, reduce((t-->[a]))).
action(2, end, reduce((t-->[a]))).
action(3, end, accept).
action(4, +, shift(7)).
action(4, ’)’, shift(6)).
action(5, +, reduce((e-->[t]))).
action(5, ’(’, reduce((e-->[t]))).
action(5, ’)’, reduce((e-->[t]))).
action(5, a, reduce((e-->[t]))).
action(5, end, reduce((e-->[t]))).
action(6, +, reduce((t-->[’(’, e, ’)’]))).
action(6, ’(’, reduce((t-->[’(’, e, ’)’]))).
action(6, ’)’, reduce((t-->[’(’, e, ’)’]))).
action(6, a, reduce((t-->[’(’, e, ’)’]))).
action(6, end, reduce((t-->[’(’, e, ’)’]))).
action(7, ’(’, shift(1)).
action(7, a, shift(2)).
action(8, +, reduce((e-->[e, +, t]))).
action(8, ’(’, reduce((e-->[e, +, t]))).
action(8, ’)’, reduce((e-->[e, +, t]))).
action(8, a, reduce((e-->[e, +, t]))).
action(8, end, reduce((e-->[e, +, t]))).
action(0, +, error).
action(0, ’)’, error).
action(0, end, error).
action(1, +, error).
action(1, ’)’, error).
action(1, end, error).
action(3, +, error).
action(3, ’(’, error).

action(3, ’)’, error).
action(3, a, error).
action(4, ’(’, error).
action(4, a, error).
action(4, end, error).
action(7, +, error).
action(7, ’)’, error).
action(7, end, error).

Die Ergebnisse stimmen mit dem oben gezeigten Beispiel nach A.Appel überein, außer daß die Zustandsnummern gemäß 1 ↔ 2, 5 ↔ 6 umzurechnen sind.

Erkennung mit dem LR(0)-Parser M_G

Der Parser erkennt passende Eingaben:

⟨Erkennung mit dem Kellerautomaten⟩≡

?- trace(parse,[call,exit]), trace(action,[exit]).
Yes
[debug] ?- parse([’(’,a,+,a,’)’]).
 T Call: (8) parse([’(’, a, +, a, ’)’])
 T Call: (9) parse([0], [’(’, a, +, a, ’)’])
 T Exit: (10) action(0, ’(’, shift(1))
 T Call: (10) parse([1, 0], [a, +, a, ’)’])
 T Exit: (11) action(1, a, shift(2))
 T Call: (11) parse([2, 1, 0], [+, a, ’)’])
 T Exit: (12) action(2, +, reduce((t-->[a])))
 T Call: (12) parse([5, 1, 0], [+, a, ’)’])
 T Exit: (13) action(5, +, reduce((e-->[t])))
 T Call: (13) parse([4, 1, 0], [+, a, ’)’])
 T Exit: (14) action(4, +, shift(7))
 T Call: (14) parse([7, 4, 1, 0], [a, ’)’])

 T Exit: (15) action(7, a, shift(2))
 T Call: (15) parse([2, 7, 4, 1, 0], [’)’])
 T Exit: (16) action(2, ’)’, reduce((t-->[a])))
 T Call: (16) parse([8, 7, 4, 1, 0], [’)’])
 T Exit: (17) action(8, ’)’, reduce((e-->[e, +, t])))
 T Call: (17) parse([4, 1, 0], [’)’])
 T Exit: (18) action(4, ’)’, shift(6))
 T Call: (18) parse([6, 4, 1, 0], [])
 T Exit: (19) action(6, end, reduce((t-->[’(’, e, ’)’])))
 T Call: (19) parse([3, 0], [])
 T Exit: (20) action(3, end, accept)

 accept
 T Exit: (19) parse([3, 0], [])
 T Exit: (18) parse([6, 4, 1, 0], [])
 T Exit: (17) parse([4, 1, 0], [’)’])
 T Exit: (16) parse([8, 7, 4, 1, 0], [’)’])
 T Exit: (15) parse([2, 7, 4, 1, 0], [’)’])
 T Exit: (14) parse([7, 4, 1, 0], [a, ’)’])
 T Exit: (13) parse([4, 1, 0], [+, a, ’)’])
 T Exit: (12) parse([5, 1, 0], [+, a, ’)’])
 T Exit: (11) parse([2, 1, 0], [+, a, ’)’])
 T Exit: (10) parse([1, 0], [a, +, a, ’)’])
 T Exit: (9) parse([0], [’(’, a, +, a, ’)’])
 T Exit: (8) parse([’(’, a, +, a, ’)’])
Yes

Beispiel 2

Die folgende Grammatik ist keine LR(0)-Grammatik.

⟨Grammatiken/lr1.grm⟩≡

term_expansion((A --> B), (A --> B)).

start_symbol(e).
e --> t, +, e.
e --> t.
t --> a.

Die Berechnung des LR(0)-Automaten A_G ergibt einen Zustand, der eine shift und eine reduce-Aktion empfiehlt:

⟨Berechnung des LR(0)-Automaten⟩≡

?- [’Grammatiken/lr1.grm’].
?- compute_lr0_automaton,
   listing(closure), listing(goto), listing(action).

closure(0, (start-->[], [e, end])).
closure(0, (e-->[], [t, +, e])).
closure(0, (e-->[], [t])).
closure(0, (t-->[], [a])).
closure(1, (t-->[a], [])).
closure(2, (start-->[e], [end])).
closure(3, (e-->[t], [+, e])).
closure(3, (e-->[t], [])).
closure(4, (e-->[t, +], [e])).
closure(4, (e-->[], [t, +, e])).
closure(4, (e-->[], [t])).
closure(4, (t-->[], [a])).
closure(5, (e-->[t, +, e], [])).

goto(0, a, 1).
goto(0, e, 2).
goto(0, t, 3).
goto(3, +, 4).
goto(4, a, 1).
goto(4, e, 5).
goto(4, t, 3).

action(0, a, shift(1)).
action(1, +, reduce((t-->[a]))).
action(1, a, reduce((t-->[a]))).
action(1, end, reduce((t-->[a]))).
action(2, end, accept).
action(3, +, reduce((e-->[t]))). % Konflikt:
action(3, +, shift(4)). % reduce oder shift ??
action(3, a, reduce((e-->[t]))).
action(3, end, reduce((e-->[t]))).
action(4, a, shift(1)).
action(5, +, reduce((e-->[t, +, e]))).
action(5, a, reduce((e-->[t, +, e]))).
action(5, end, reduce((e-->[t, +, e]))).
action(0, +, error).
action(0, end, error).
action(2, +, error).
action(2, a, error).
action(4, +, error).
action(4, end, error).

Wegen des Konflikts im Zustand 3 ist die nächste Aktion nicht eindeutig bestimmt, die Grammatik also keine LR(0)-Grammatik.